code:
;-------------------------------------------------
;진찰을 받는다
;외출 커맨드
;-------------------------------------------------
@COM616
PRINTL 「오늘은 무슨 일일까?」
$LOOP
DRAWLINE
PRINTL [0] 피로가 풀리지 않아서…
PRINTL [1] 왠지 모르게 피곤해지고…
;SIF MAXBASE:MASTER:정력 - BASE:MASTER:정력 > 500
;	PRINTL [2] 그녀가 격렬해서…
PRINTL [3] 요즘 그녀와 매너리즘에 빠져서…
SIF TALENT:MASTER:청소중독
	PRINTL [4] 청소가 너무 즐거워서 그만둘 수가 없습니다
;PRINTL [5] 선생님의 일을 생각하면(자) 자지가 쑥갓응입니다
;PRINTL [5] 선생님의 일을 생각하면(자) 가랑이가 근질근질 합니다
;SIF GETBIT(TALENT:MASTER:성별 & 1)
;PRINTL [10] 자지의 고민
DRAWLINE
PRINTL [100] 그만둔다
INPUT
IF RESULT == 0
	CALL REMEDY(RAND:3)
	PRINTL 치료비로 \3000 청구되었습니다
	IF MONEY < 3000
		DOWNBASE:MASTER:체력 -= 100
		DOWNBASE:MASTER:기력 -= 200
		TIME += 180
		TFLAG:193 = 1
	ELSE
		MONEY -= 3000
	ENDIF
ELSEIF RESULT == 1
	;에이린의 사모 연모로 양화
	CALL REMEDY(RAND:2 + 3 + TALENT:(NO:72):3 * 2 + TALENT:(NO:72):8)
	PRINTL 치료비로 \3000 청구되었습니다
	IF MONEY < 3000
		DOWNBASE:MASTER:체력 -= 100
		DOWNBASE:MASTER:기력 -= 200
		TIME += 180
		TFLAG:193 = 1
	ELSE
		MONEY -= 3000
	ENDIF
ELSEIF RESULT == 3
	PRINTL 「이런 건 어떨까」
	DRAWLINE
	$SELL
	FOR LOCAL, 12,17
		SIF ITEMNAME:LOCAL == ""
			CONTINUE
		PRINTFORM [{LOCAL}] %ITEMNAME:LOCAL% ￥{(ITEMPRICE:LOCAL) / 5}
		IF ITEM:LOCAL
			PRINTL 　(매진)
		ELSE
			PRINTL
		ENDIF
	NEXT
	PRINTL
	PRINTL [100] 돌아온다
	$LOOP2
	INPUT
	IF RESULT >= 12 && RESULT <= 17
		CALL BUY_ABLE(RESULT, 0,1,80)
		GOTO SELL
	ELSEIF RESULT == 100
		RETURN 0
	ELSE
		GOTO LOOP2
	ENDIF
ELSEIF RESULT == 4 && TALENT:MASTER:청소중독
	IF FLAG:850 == 0
		PRINTL 「집안일을 좋아해? 좋은 일 아냐」
		PRINTL 「그런 게 아니라고?」
		PRINTW 「다른 일이 손에 잡히지 않아? 그건 곤란하군」
		PRINTW …
		PRINTW 「으―음, 역시 좀 이상하네」
		PRINTW ……
		PRINTW 「성적으로 흥분!?」
		PRINTW …………
		PRINTL 「도대체 뭘 하면 이렇게 되는 거야」
		PRINTL 「희귀한 증례를 보여 주었으니까 이번은 공짜로 해도 좋아」
		PRINTL 「앞으로는 이상한 청소는 하지 말도록」
		PRINTFORMW %CALLNAME:MASTER%의 [청소중독]이 나았다
		TALENT:MASTER:청소중독 = 0
		FLAG:850 ++
	ELSE
		PRINTL 「또 그걸 재발시켰다고?」
		PRINTFORML 「정말이지… 이번에는 ￥{FLAG:850 * 10000} 받을게」
		IF MONEY < FLAG:850 * 10000
			PRINTL 「치료비가 부족한가 보네」
			PRINTL 「돈이 모이면 다시 와」
		ELSE
			PRINTW …
			PRINTW ……
			PRINTW …………
			PRINTL 「네, 끝」
			PRINTL 「또 재발시키는 일이 있으면 더 많이 지불하도록 할거니깐」
			PRINTFORMW %CALLNAME:MASTER%의 [청소중독]이 나았다
			TALENT:MASTER:청소중독 = 0
			MONEY -= FLAG:850 * 10000
			FLAG:850 ++
		ENDIF
	ENDIF
ELSEIF RESULT == 6
	IF GETBIT(CFLAG:72:기정사실, 1) && EXP:72:펠라티오경험 > 10 && !TALENT:72:키스미경험
		IF TALENT:72:기분 == -1
			PRINTFORMDW %조사처리(CALLNAME:72,"는")% 한숨을 쉬자 %마스터를% 안쪽으로 안내했다
			PRINTFORMDW %조사처리(CALLNAME:72,"는")% 기분나쁜 듯 옆머리를 쓸어올리면서 %CALLNAME:MASTER%의 음경을 물어 넣었다
			PRINTFORMDW %조사처리(CALLNAME:72,"는")% 정액을 입안에 받아낸 후 양치질을 하고 손을 씻어 진찰하러 돌아갔다
			MAXBASE:MASTER:정력 +=1*3;;정력증가
			EXP:MASTER:사정경험 ++
			BASE:MASTER:정력 = MAX(BASE:MASTER:정력 - 200, 0)
			EXP:72:펠라티오경험 ++
			EXP:72:정액경험 ++
			IF TALENT:MASTER:농후정액
				PRINTFORMW %CALLNAME:MASTER%의 [농후정액]이 나았다
				TALENT:MASTER:농후정액 = 0
			ENDIF
			PRINTL 치료비로 \10000 청구되었습니다
			MONEY -= 10000
		ELSE
			PRINTFORMDW 에이린에게 바쁘니까 다음에라며 가볍게 다루어져 버렸다…
			TALENT:72:기분 --
		ENDIF
	ELSE
		PRINTFORMDW %조사처리(CALLNAME:72,"는")% 감사의 말을 하며 %마스터를% 안쪽으로 안내했다
		PRINTW …
		PRINTW ……
		PRINTW ………
		PRINTFORMDW %마스터는% 헌혈에 협력하게 되었다…
		BASE:MASTER:체력 = MIN(MAXBASE:MASTER:체력, BASE:MASTER:체력 - 400)
		TALENT:72:기분 = 1
	ENDIF
;ELSEIF RESULT == 10
;	CALL PENIS_CHANGE
ELSEIF RESULT == 100
	RETURN -1
ELSE
	GOTO LOOP
ENDIF
SETBIT TFLAG:일일일회, 3
TIME += 30
RETURN 1

@REMEDY(ARG)
SELECTCASE ARG
CASE 0
	LOCALS = 맛있는 밥
	BASE:MASTER:체력 = MIN(MAXBASE:MASTER:체력, BASE:MASTER:체력 + 500)
CASE 1
	LOCALS = 영양의 주사
	BASE:MASTER:체력 = MIN(MAXBASE:MASTER:체력, BASE:MASTER:체력 + 750)
CASE 2
	LOCALS = 불끈불끈Z
	BASE:MASTER:체력 = MAXBASE:MASTER:체력
CASE 3
	LOCALS = 계절원―
	BASE:MASTER:기력 = MIN(MAXBASE:MASTER:기력, BASE:MASTER:기력 + 500)
CASE 4
	LOCALS = 진정의 주사
	BASE:MASTER:기력 = MIN(MAXBASE:MASTER:기력, BASE:MASTER:기력 + 750)
CASE 5
	LOCALS = 이마로 체온
	BASE:MASTER:기력 = MIN(MAXBASE:MASTER:기력, BASE:MASTER:기력 + 900)
CASE 6
	LOCALS = 껴안는다
	BASE:MASTER:기력 = MIN(MAXBASE:MASTER:기력, BASE:MASTER:기력 + 1000)
ENDSELECT

PRINTFORMW 에이린은 %LOCALS%에 의한 치료를 실시했다
SIF ARG == 2
	PRINTL 후아 싫어해 개, 만 타 응!
SIF ARG >= 0 && ARG <= 2
	PRINTW 체력이 회복했다
SIF ARG >= 3 && ARG <= 6
	PRINTW 기력이 회복했다
[SKIPSTART]
@PENIS_CHANGE
$TINKO_LOOP
PRINTL [1] 동심에 돌아가고 싶기 때문에 아이 같은 자지로 해 주세요
PRINTL [2] 방탕에 빠지지 않게 데츄 해 주세요
PRINTL [3] 보통 자지로 해 주세요
PRINTL [4] 밤길을 걸을 때 무서워서 빛나는 듯하게 해 주세요
PRINTL [5] 여자아이를 히-히-말하게 하고 싶어서 진주 넣어 주세요
PRINTL [6] 자신을 가질 수 있도록(듯이) 크게 해 주세요
;PRINTL [7] 오크를 동경하고 있기 때문에 드릴같이 해 주세요 그리고 엘프를 찾으러 갑니다
;PRINTL [8] 피임이라든지 귀찮아 응으로 정관 수술 부탁합니다
INPUT
SIF TALENT:MASTER:형상 == RESULT
	GOTO TINKO_LOOP
	LOCAL = RESULT
SELECTCASE RESULT
	CASE 1
		IF GETBIT(CFLAG:72:기정사실, 1) && CFLAG:72:2 > 5000
			PRINTFORMW 에이린은 그러한 것도 개미일까와 생각한 것 같은…
		ELSE
			PRINTFORMW 에이린은 경직된 얼굴로 진심인지 (들)물어 왔다…
	CASE 2
		IF GETBIT(CFLAG:72:기정사실, 1) && CFLAG:72:2 > 5000
			PRINTFORMW 에이린은 다시 생각하도록(듯이) 갔다왔다…
		ELSE
	CASE 3
		PRINTFORMW 에이린은 보통이 제일이라면 기분 좋게 맡아 주었다
	CASE 4
		PRINTFORMW 에이린은 이성인과 컨택 하는 듯한 상태로, 아무것도 그런 곳을 번뜩거릴 수 없어도 괜찮지 않을까라고 (들)물어 왔다…
	CASE 5
		PRINTFORMW 에이린은 그런 것은
[SKIPEND]

;-------------------------------------------------
;실행 가능 판정
;-------------------------------------------------
;진찰
@COM_ABLE616
;일괄관리
SIF GLOBAL_COMABLE(616)
	RETURN RESULT
;에이린이 없으면 안 돼
SIF CFLAG:MASTER:현재위치 != CFLAG:72:현재위치
	RETURN 0
;데이트중
SIF CHK_DATENOW(CFLAG:TARGET:데이트중) && TARGET > 0
	RETURN 0
;데이트 도중
SIF (CFLAG:현재위치 % 100) % 10
	RETURN 0
SIF CFLAG:MASTER:현재위치 != 영원정
	RETURN 0
SIF GETBIT(TFLAG:일일일회, 3)
	RETURN 0
;수면중
SIF CFLAG:MASTER:수면
	RETURN 0
;시간정지중
SIF FLAG:70
	RETURN 0
RETURN 1

path: ERB/커맨드관련/COMF/외출계/COMF616 진찰받는다_t.ERB


code:
;-------------------------------------------------
;과자가게에 권한다
;외출 커맨드
;-------------------------------------------------
@COM617
#DIM 회복전체력1
#DIM 회복전기력1
#DIM 회복후체력1
#DIM 회복후기력1
#DIM 회복전체력2
#DIM 회복전기력2
#DIM 회복후체력2
#DIM 회복후기력2

회복전체력1 = BASE:MASTER:체력
회복전기력1 = BASE:MASTER:기력
회복전체력2 = BASE:TARGET:체력
회복전기력2 = BASE:TARGET:기력
LOCAL = RAND:100
LOCAL:1 = 90 + GET_SUCCESS_RATE() / 5
SIF LOCAL:1 > 99
	LOCAL:1 = 99
IF LOCAL <= 9
	;대성공 ;붉은실 KR판 오리지날
	CALL 붉은실발동
	EXP:MASTER:회화경험 ++
	EXP:MASTER:데이트경험 ++
	EXP:TARGET:데이트경험 ++
ELSEIF LOCAL >= LOCAL:1
	;실패
	TFLAG:193 = -1
ELSE
	;성공
	TFLAG:193 = 0
ENDIF
TFLAG:98 += 1

DOWNBASE:기력 += 10

;고정으로 획득하는 소스
SOURCE:환락 = 800
SOURCE:사랑 = 300

;ABL:욕망을 본다
IF ABL:욕망 <= 1
	SOURCE:환락 += (ABL:욕망 * 60)
ELSEIF ABL:욕망 <= 3
	SOURCE:환락 += 300 + (ABL:욕망 * 60)
ELSEIF ABL:욕망 <= 5
	SOURCE:환락 += 600 + (ABL:욕망 * 60)
ELSEIF ABL:욕망 <= 8
	SOURCE:환락 += 900 + (ABL:욕망 * 75)
ELSEIF ABL:욕망 <= 11
	SOURCE:환락 += 1500 + (ABL:욕망 * 75)
ELSE
	SOURCE:환락 += 2500 + (ABL:욕망 * 30)
ENDIF

;ABL:봉사정신을 본다
IF ABL:봉사정신 <= 1
	SOURCE:사랑 += (ABL:봉사정신 * 30)
ELSEIF ABL:봉사정신 <= 3
	SOURCE:사랑 += 200 + (ABL:봉사정신 * 30)
ELSEIF ABL:봉사정신 <= 5
	SOURCE:사랑 += 500 + (ABL:봉사정신 * 30)
ELSEIF ABL:봉사정신 <= 8
	SOURCE:사랑 += 700 + (ABL:봉사정신 * 45)
ELSEIF ABL:봉사정신 <= 11
	SOURCE:사랑 += 1200 + (ABL:봉사정신 * 45)
ELSE
	SOURCE:사랑 += 1800 + (ABL:봉사정신 * 20)
ENDIF

;호감 도를 본다
IF CFLAG:2 <= 500
	SOURCE:환락 += CFLAG:2
ELSEIF CFLAG:2 <= 5000
	SOURCE:환락 += 750 + (CFLAG:2 - 500) / 3
ELSE
	SOURCE:환락 += 2000 + (CFLAG:2 - 5000) / 4
ENDIF
SIF SOURCE:환락 < 0
	SOURCE:환락 = 0

SOURCE:수동 = 200 + 100 * ABL:순종
SOURCE:정복 = 200 + 100 * ABL:새드끼

IF TFLAG:193 == -1
	TIMES SOURCE:환락 , 0.10
	TIMES SOURCE:사랑 , 0.10
	TIMES SOURCE:정복 , 0.50
	TIMES SOURCE:수동 , 0.50
ELSEIF TFLAG:193 == 0
	TIMES SOURCE:환락 , 1.00
	TIMES SOURCE:사랑 , 1.00
	TIMES SOURCE:정복 , 1.00
	TIMES SOURCE:수동 , 1.00
ELSE
	TIMES SOURCE:환락 , 2.00
	TIMES SOURCE:사랑 , 1.50
	TIMES SOURCE:정복 , 2.00
	TIMES SOURCE:수동 , 2.00
	SIF TALENT:15 < 1
		TALENT:15 = 1
ENDIF

BASE:체력 = MIN(BASE:체력 + MAXBASE:체력 / 25 , MAXBASE:체력)
BASE:기력 = MIN(BASE:기력 + MAXBASE:기력 / 4 , MAXBASE:기력)
BASE:MASTER:체력 = MIN(BASE:MASTER:체력 + MAXBASE:MASTER:체력 / 25 , MAXBASE:MASTER:체력)
BASE:MASTER:기력 = MIN(BASE:MASTER:기력 + MAXBASE:MASTER:기력 / 4 , MAXBASE:MASTER:기력)

TIME += 60
CALL 붉은실발동
EXP:MASTER:데이트경험 ++
EXP:TARGET:데이트경험 ++
CALL 만복도상승(TARGET,"디저트")
CALL 만복도상승(MASTER,"디저트")
MONEY -= 3000
회복후체력1 = BASE:MASTER:체력
회복후기력1 = BASE:MASTER:기력
회복후체력2 = BASE:TARGET:체력
회복후기력2 = BASE:TARGET:기력
DRAWLINE
PRINTFORML 체력+{ABS(회복후체력1 - 회복전체력1)}(%CALLNAME:MASTER%)
PRINTFORML 기력+{ABS(회복후기력1 - 회복전기력1)}(%CALLNAME:MASTER%)
PRINTFORML 체력+{ABS(회복후체력2 - 회복전체력2)}(%CALLNAME:TARGET%)
PRINTFORML 기력+{ABS(회복후기력2 - 회복전기력2)}(%CALLNAME:TARGET%)
RETURN 1

;-------------------------------------------------
;실행 가능 판정
;-------------------------------------------------
;과자가게로 이끈다
@COM_ABLE617
;일괄관리
SIF GLOBAL_COMABLE(617)
	RETURN RESULT
;상대가 없다
SIF !TARGET
	RETURN 0
;데이트중 이외
SIF !CHK_DATENOW(CFLAG:12)
	RETURN 0
;甘味処ではない場所
SIF !SWEETS_SHOP(CFLAG:MASTER:현재위치)
	RETURN 0
;먹었던 바로 직후
SIF !TIME_PROGRESS(TCVAR:디저트공복시각)
	RETURN 0
SIF MONEY < 3000
	RETURN 0
;수면중
SIF CFLAG:수면
	RETURN 0
;시간정지중
SIF FLAG:70
	RETURN 0
RETURN 1

;甘味処を判定する関数
@SWEETS_SHOP(ARG)
#FUNCTION
IF !IN_HOME(ARG)
	SIF ARG == 상가거리
		RETURNF 1
ELSE
	;甘味所
	SIF ARG == 217
		RETURNF 1
ENDIF
RETURNF 0

path: ERB/커맨드관련/COMF/외출계/COMF617 과자가게로부른다_t.ERB

license: mit


code:
;-------------------------------------------------
;꽃집
;외출 커맨드
;-------------------------------------------------
@COM618
PRINTL 소지 아이템
CALL LIST_ITEM, "야채"
CALL LIST_ITEM, "원예용품"
PRINTL
PRINTL
PRINTFORML 구입할 아이템을 선택해주세요 소지금:\\{MONEY}

CALL SHOW_ITEM, "원예용품"
PRINT [100] - 돌아간다
$INPUT_LOOP
INPUT
IF RESULT == 100
	TIME += 10
	RETURN -1
ELSEIF ITEM_TYPES( RESULT, "원예용품" )
	IF RESULT == 150 && ITEM:원예용품일식
		PRINTW 이미 가지고 있습니다
	ELSE
		CALL BUY_ABLE, RESULT
	ENDIF
	PRINTL
	RESTART
ELSE
	GOTO INPUT_LOOP
ENDIF
TIME += 10

;-------------------------------------------------
;실행 가능 판정
;-------------------------------------------------
;꽃집
@COM_ABLE618
;일괄관리
SIF GLOBAL_COMABLE(618)
	RETURN RESULT

;花屋ではない場所
SIF !FLOWER_SHOP(CFLAG:MASTER:현재위치)
	RETURN 0
;수면중
SIF CFLAG:수면
	RETURN 0
;시간정지중
SIF FLAG:70
	RETURN 0
RETURN 1

;花屋を判定する関数
@FLOWER_SHOP(ARG)
#FUNCTION
;花屋
SIF ARG == GET_MAP_REPLACEMENT(215)
	RETURNF 1
RETURNF 0

path: ERB/커맨드관련/COMF/외출계/COMF618 꽃집_t.ERB


code:
;-------------------------------------------------
;술집
;외출 커맨드
;-------------------------------------------------
@COM619
#DIM 탁주판매개수
#DIM 청주판매개수
#DIM 오니고로시판매개수
#DIM 재고증가수
;RAND를 걸치는 것은 일일일회
IF FLAG:마지막에술집에방문한날 != DAY
	;연회 당일은 술이 품귀상태로
	IF FLAG:개시일 == DAY
		탁주판매개수 = RAND:(4 + FLAG:술집매각율플래그 * 4)
		청주판매개수     = RAND:(3 + FLAG:술집매각율플래그 * 3)
		오니고로시판매개수   = RAND:(2 + FLAG:술집매각율플래그 * 2)
	ELSE
		탁주판매개수 = RAND:(6 + FLAG:술집매각율플래그 * 8)
		청주판매개수     = RAND:(5 + FLAG:술집매각율플래그 * 6)
		오니고로시판매개수   = RAND:(4 + FLAG:술집매각율플래그 * 4)
	ENDIF
	;누적 구입액에 응해도 재고가 증가
	IF FLAG:술집에서누적구입액
		재고증가수 = MAX(FLAG:술집에서누적구입액 / 100000 - 2, 0) + 1
		탁주판매개수 += RAND:재고증가수
		청주판매개수     += RAND:재고증가수
		오니고로시판매개수   += RAND:재고증가수
	ENDIF
ENDIF

IF FLAG:마지막에술집에방문한날 == 0
	PRINTL 「첫 손님이구나? 서비스할게」
ELSEIF DAY - FLAG:마지막에술집에방문한날 > 7
	PRINTL 「오랜만이네요―, 천천히 둘러봐요」
ELSEIF FLAG:마지막에술집에방문한날 == DAY
	PRINTL 「어서 오세요, 응, 어머머? 웬 일이야?」
ELSEIF FLAG:술집에서누적구입액 > 1000000 && FLAG:개시일 != DAY
	PRINTFORML 「어서 오세요, %마스터를% 위해서 널럴하게 들여놨지」
ELSEIF FLAG:개시일 == DAY
	PRINTL 「때가 안 좋았네. 아까 온 손님이 거의 다 사갔어」
ELSEIF 탁주판매개수 + 청주판매개수 + 오니고로시판매개수 > 30
	PRINTL 「어서 오세요, 좋은 술이 많이 있어―」
ELSE
	PRINTL 「어서 오십시오―」
ENDIF
SIF FLAG:술집매입강화
	PRINTFORML 「현재 매입 강화 캠페인 중이야. 물건 조달해주면 톡톡히 쳐줄게」
$INPUT_LOOP2
PRINTL [0] - 술을 산다
PRINTL [1] - 술을 판다
PRINTL [100] - 돌아간다
$INPUT_LOOP
INPUT

IF RESULT == 0
	CALL 술구입(탁주판매개수, 청주판매개수, 오니고로시판매개수)
	GOTO INPUT_LOOP2
ELSEIF RESULT == 1
	CALL 술도매(탁주판매개수, 청주판매개수, 오니고로시판매개수)
	GOTO INPUT_LOOP2
ELSEIF RESULT != 100
	GOTO INPUT_LOOP
ENDIF

FLAG:마지막에술집에방문한날 = DAY
IF FLAG:술집누적전매이익 > 3000000 && FLAG:술집매각율플래그 == 2
	FLAG:술집매각율플래그 = 3
	PRINTL
	PRINTL 가게에서 나가려 했는데 불러 세웠다
	PRINTL 「두둑하게 술을 조달해줘서 정말 살겠어∼, 앞으로는 매입 금액 더 쳐줄게」
ELSEIF FLAG:술집누적전매이익 > 1000000 && FLAG:술집매각율플래그 == 1
	FLAG:술집매각율플래그 = 2
	PRINTL
	PRINTL 가게에서 나가려 했는데 불러 세웠다
	PRINTL 「두둑하게 술을 조달해줘서 살겠어―, 매입 금액 더 쳐줄 테니 더 조달해주지 않을래?」
ELSEIF FLAG:술집누적전매이익 > 100000 && FLAG:술집매각율플래그 == 0
	FLAG:술집매각율플래그 = 1
	PRINTL
	PRINTL 가게에서 나가려 했는데 불러 세웠다
	PRINTL 「늘 술을 조달해줘서 고마워. 사례로 다음 번부터 매입 금액 더 쳐줄게―」
ELSE
	PRINTL 「또 오세요―」
ENDIF
TIME += 10
RETURN -1

@술구입(탁주판매개수, 청주판매개수, 오니고로시판매개수)
#DIM REF 탁주판매개수
#DIM REF 청주판매개수
#DIM REF 오니고로시판매개수
#DIM 판매개수
#DIM 소지가능개수
#DIM 가격인하율
#DIM 선택한상품
#DIM 구입개수
#DIM 분명하게샀다

;가격인하율을 20%로 설정(정가의 80%로 판매)
가격인하율 = 20

;각각의 술의 최대 소지 가능수를 80으로 설정(80개 이상 가지고 있다고(면) 그 술은 살 수 없습니다)
소지가능개수 = 80

PRINTL
IF FLAG:마지막에술집에방문한날 == 0
	PRINTL 「처음 왔으니 서비스야―」
	;가격인하율을 40%로 설정(정가의 60%로 판매)
	가격인하율 = 40
ELSE
	PRINTL 「어느 거나 사서 손해 없어―」
ENDIF
$INPUT_LOOP2
PRINTFORML 소지금 :\\%TOSTR(MONEY,"#,###")%
FOR LOCAL, 51,54
	SELECTCASE LOCAL
		CASE 51
			판매개수 = 탁주판매개수
		CASE 52
			판매개수 = 청주판매개수
		CASE 53
			판매개수 = 오니고로시판매개수
	ENDSELECT
	LOCALS = [{LOCAL}]%ITEMNAME:(LOCAL), 8, LEFT%	(\\%TOSTR(ITEMPRICE:(LOCAL) * (100 - 가격인하율) / 100), 5, RIGHT%)
	IF ITEMPRICE:LOCAL
		PRINTFORM %LOCALS%
		IF 판매개수 == 0
			SETCOLOR C_RED
			PRINT 	(다 팔림)
			RESETCOLOR
		ELSE
			PRINTFORM 	(재고:{판매개수})
		ENDIF
		IF ITEM:LOCAL >= 소지가능개수
			SETCOLOR C_YELLOW
			PRINTFORM \t소지:{ITEM:LOCAL, 2, RIGHT}
			FONTSTYLE 1
			PRINTL 구입 불가
			RESETCOLOR
			FONTREGULAR
		ELSE
			PRINTFORML \t소지:{ITEM:LOCAL, 2, RIGHT}
		ENDIF
	ENDIF
NEXT
PRINT [100] - 돌아간다
$INPUT_LOOP
INPUT
IF RESULT == 100
	RETURN -1
ELSEIF RESULT >= 51 && RESULT <= 53

	;나중에 다시 한번 INPUT를 사용하기 위해 지금의 값을 퇴피
	선택한상품 = RESULT

	;선택한 번호에 따라판매개수를 격납
	SELECTCASE 선택한상품
		CASE 51
			판매개수 = 탁주판매개수
		CASE 52
			판매개수 = 청주판매개수
		CASE 53
			판매개수 = 오니고로시판매개수
	ENDSELECT

	;품절의 경우
	IF 판매개수 == 0
		PRINTL
		PRINTL 「미안해요. 다 팔렸어요」

	;최대 소지 가능수이상의 술을 소지하고 있는 경우
	ELSEIF ITEM:선택한상품 >= 소지가능개수
		PRINTL 「손님, 그만큼 있으면 충분하잖아요? 오히려 우리한테 물건을 대주라―」

	;그 이외의경우 개수를 묻는다
	ELSE
		구입개수 = 0
		$BUY_LOOP
		PRINTFORML %ITEMNAME:선택한상품%를 몇 개 살까…
		PRINTFORML 사는 수：{구입개수}
		PRINTFORML [ -10]　[  -1]　[  +1]　[ +10]
		PRINTBUTTON "[전부] ", 999
		PRINTBUTTON " [리셋]", -999
		PRINTL
		PRINTBUTTON "[산다]", 0
		INPUT
		IF RESULT
			IF RESULT == 999
				구입개수 =판매개수
			ELSEIF RESULT == -999
				구입개수 = 0
			ELSE
				구입개수 = LIMIT(구입개수 + RESULT, 0,판매개수)
			ENDIF
			CLEARLINE 6
			GOTO BUY_LOOP
		ENDIF
		;0이하의 값을 입력했을 경우
		IF 구입개수 <= 0
			PRINTL
			PRINTW 역시 사는 건 그만두자
			GOTO INPUT_LOOP2
		;재고의 수보다 많이 구입하려고 했을 경우
		ELSEIF 판매개수 < 구입개수
			PRINTW 재고가 그렇게 많진 않은 것 같다
			GOTO INPUT_LOOP2
		;최대 소지 가능수보다 많이 구입하려고 했을 경우
		ELSEIF 소지가능개수 < ITEM:선택한상품 + 구입개수
			PRINTW 그렇게나 가져봤자 처치곤란하다
			GOTO INPUT_LOOP2
		ENDIF
		CALL BUY_ABLE(선택한상품, 0, 구입개수, 가격인하율)
		분명하게샀다 = RESULT
		IF FLAG:술집에서누적구입액 > 100000 && !RAND:9 && 분명하게샀다
			PRINTL
			PRINTL 「항상 사줘서 고마워, 이건 덤이야」
			PRINTFORMW ―덤으로 %ITEMNAME:510%를 받았다-
			ITEM:510 ++
		ELSE
			PRINTL
			PRINTW 「매번 감사―」
		ENDIF
		IF 분명하게샀다
			FLAG:술집에서누적구입액 += ITEMPRICE:선택한상품 * (100 - 가격인하율) / 100 * 구입개수
			;판매개수를 감소시키는 처리
			SELECTCASE 선택한상품
				CASE 51
					탁주판매개수 -= 구입개수
				CASE 52
					청주판매개수     -= 구입개수
				CASE 53
					오니고로시판매개수   -= 구입개수
			ENDSELECT
		ENDIF
	ENDIF
	GOTO INPUT_LOOP2
ELSE
	GOTO INPUT_LOOP2
ENDIF
RETURN

@술도매(탁주판매개수, 청주판매개수, 오니고로시판매개수)
#DIM REF 탁주판매개수
#DIM REF 청주판매개수
#DIM REF 오니고로시판매개수
#DIM 가격인하율
#DIM 소지수
#DIM 선택한상품
#DIM 매각개수
#DIM 매각액

SELECTCASE FLAG:술집매각율플래그
	CASE 0
		;가격인하율을 90%로 설정(정가의 10%로 구입해 준다)
		가격인하율 = 90
	CASE 1
		;가격인하율을 85%로 설정(정가의 15%로 구입해 준다)
		가격인하율 = 85
	CASE 2
		;가격인하율을 80%로 설정(정가의 20%로 구입해 준다)
		가격인하율 = 80
	CASE 3
		;가격인하율을 70%로 설정(정가의 30%로 구입해 준다)
		가격인하율 = 70
ENDSELECT
SIF FLAG:술집매입강화
	가격인하율 -= 20
PRINTL
PRINT 「
IF FLAG:술집매입강화
	PRINT 매입 강화 캠페인중이니까 특별히
ELSEIF FLAG:술집매각율플래그
	PRINT 고객이니까 특별히
ENDIF
PRINTFORML 정가의 {100 - 가격인하율}\%로 살게」
PRINTL 무엇을 팔까…
$INPUT_LOOP2
PRINTFORML 소지금 :\\%TOSTR(MONEY,"#,###")%
FOR LOCAL, 51,54
	LOCALS = [{LOCAL}]%ITEMNAME:(LOCAL)%
	IF ITEMPRICE:LOCAL
		PRINTFORML %LOCALS%	(\\{ITEMPRICE:(LOCAL) * (100 - 가격인하율) / 100})  	(소지:{ITEM:LOCAL})
	ENDIF
NEXT
PRINT [100] - 돌아간다
$INPUT_LOOP
INPUT
IF RESULT == 100
	RETURN -1
ELSEIF RESULT >= 51 && RESULT <= 53
	;선택한 번호에 따라 소지수를 격납
	소지수 = ITEM:RESULT

	;소지 0의 경우
	IF 소지수 == 0
		PRINTL
		PRINTL 물건이 없기에 팔고 싶어도 팔 수 없다
	;소지 0으로 없는 경우 개수를 묻는다
	ELSE
		;다시 한번 INPUT를 사용하기 때문에 지금의 값을 퇴피
		선택한상품 = RESULT
		매각개수 = 0
		$SELL_LOOP
		PRINTFORML %ITEMNAME:선택한상품%를 몇 개 팔까…
		PRINTFORML 파는 수：{매각개수}
		PRINTFORML [-100]　[ -10]　[  -1]　[  +1]　[ +10]　[+100]
		PRINTBUTTON "[전부] ", 999
		PRINTBUTTON " [리셋]", -999
		PRINTL
		PRINTBUTTON "[판다]", 0
		INPUT
		IF RESULT
			IF RESULT == 999
				매각개수 = 소지수
			ELSEIF RESULT == -999
				매각개수 = 0
			ELSE
				매각개수 = LIMIT(매각개수 + RESULT, 0, 소지수)
			ENDIF
			CLEARLINE 6
			GOTO SELL_LOOP
		ENDIF
		IF 매각개수 <= 0
			PRINTL
			PRINTL 역시 파는 건 그만두자
			GOTO INPUT_LOOP2
		ENDIF
		IF 매각개수 > 소지수
			PRINTL
			PRINTL 그렇게나 많이 가지고 있지 않다
			GOTO INPUT_LOOP2
		ENDIF

		;1개당의 매각값을 산출
		매각액 = ITEMPRICE:선택한상품 * (100 - 가격인하율) / 100
		ITEM:선택한상품 -= 매각개수
		MONEY += 매각액 * 매각개수
		;가게는 정가의 80%로 판매
		FLAG:술집누적전매이익 += ITEMPRICE:선택한상품 * (가격인하율 - 20) / 100 * 매각개수
		PRINTFORMW %ITEMNAME:선택한상품%를 ￥{매각액*매각개수}로 {매각개수}개 매각했습니다
		;판매개수를 증가시키는 처리
		SELECTCASE 선택한상품
		CASE 51
			탁주판매개수 += 매각개수
		CASE 52
			청주판매개수     += 매각개수
		CASE 53
			오니고로시판매개수   += 매각개수
		ENDSELECT
	ENDIF
	GOTO INPUT_LOOP2
ELSE
	GOTO INPUT_LOOP
ENDIF
;-------------------------------------------------
;실행 가능 판정
;-------------------------------------------------
;술집
@COM_ABLE619
;일괄관리
SIF GLOBAL_COMABLE(619)
	RETURN RESULT
;酒屋ではない場所
SIF !SAKE_SHOP(CFLAG:MASTER:현재위치)
	RETURN 0
;수면중
SIF CFLAG:수면
	RETURN 0
;시간정지중
SIF FLAG:70
	RETURN 0
RETURN 1

;酒屋を判定する関数
@SAKE_SHOP(ARG)
#FUNCTION
;酒屋
SIF ARG == GET_MAP_REPLACEMENT(211)
	RETURNF 1
RETURNF 0

path: ERB/커맨드관련/COMF/외출계/COMF619 술집_t.ERB


code:
;대본가게
@COM620
#DIM CHARISMA
#DIM ItemId

PRINTFORML 소지금:\\{MONEY}
PRINTL 「임차료는 모두\3000 입니다」
$C_LOOP
FOR LOCAL,1, BookTitles + 1
	ItemId = LOCAL + 100
	PRINTFORM [{LOCAL}] %ITEMNAME:ItemId%
	IF ITEM:ItemId
		PRINTFORML (%마스터가% 빌리고 있다)
	ELSEIF GETBIT (FLAG:1011,LOCAL)
		PRINTL (대출중)
	ELSE
		PRINTL
	ENDIF
NEXT

IF !CHARISMA
	PRINTFORML [100] 카리스마 지불
ELSE
	PRINTFORML [100] 현금지불
ENDIF
DRAWLINE
PRINTL [0] 그만둔다

$LOOP
INPUT
FOR LOCAL,1, BookTitles + 1
	ItemId = LOCAL + 100
	IF ITEM:ItemId
		PRINTL 「2권 이상의 대출은 할 수 없습니다」
		RETURN -1
	ENDIF
NEXT

IF RESULT == 100
	IF !CHARISMA
		PRINTFORMW 「카리스마로 지불해???」
		PRINTFORMW 「그러니까……그럼 10 카리스마? 로 괜찮아……」
		IF MONEY:2 < 10
			PRINTFORMW 카리스마가 부족하다
			GOTO C_LOOP
		ENDIF
	ELSE
		PRINTFORMW 「아, 돈으로 지불하는거구나」
	ENDIF
	PRINTFORML
	CHARISMA = !CHARISMA
	GOTO C_LOOP
ELSE
	IF INRANGE(RESULT, 1, BookTitles)
		IF GETBIT (FLAG:1011,RESULT)
			PRINTL 「대출중입니다」
		ELSE
			SETBIT FLAG:1011,RESULT
			IF !CHARISMA
				MONEY -= 3000
			ELSE
				MONEY:2 -= 10
			ENDIF
			ITEM:(100 + RESULT) = 1
			FLAG:대본반환기한 = 8
			TFLAG:193 = 1
			PRINTFORML %조사처리(ITEMNAME:(100 + RESULT),"를")% 빌렸다
			RETURN 1
		ENDIF
	ELSEIF RESULT == 0
		RETURN -1
	ELSE
		GOTO LOOP
	ENDIF
ENDIF
TIME += 30

;대본소
@COM_ABLE620
;일괄관리
SIF GLOBAL_COMABLE(620)
	RETURN RESULT
;貸本屋ではない場所
SIF !BOOK_RENTAL(CFLAG:MASTER:현재위치)
	RETURN 0
SIF MONEY < 3000
	RETURN 0
;수면중
SIF CFLAG:수면
	RETURN 0
;시간정지중
SIF FLAG:70
	RETURN 0
RETURN 1

;貸本屋を判定する関数
@BOOK_RENTAL(ARG)
#FUNCTION
;鈴奈庵
SIF ARG == GET_MAP_REPLACEMENT(213)
	RETURNF 1
RETURNF 0

@BOOKLEND
FOR LOCAL,1,BookTitles + 1
	SIF LOCAL == 6	;[그림연극]을 패스
		CONTINUE
	SIF !ITEM:(100 + LOCAL) && !RAND:3
		INVERTBIT FLAG:1011,LOCAL
NEXT
SIF FLAG:대본반환기한
	FLAG:대본반환기한 -= 1
IF FLAG:대본반환기한 == 1
	FOR LOCAL,1,BookTitles + 1
		ITEM:(100 + LOCAL) = 0
	NEXT
	PRINTL 반환 기한이 왔으므로, 빌리고 있던 책을 돌려주었습니다.
ENDIF

path: ERB/커맨드관련/COMF/외출계/COMF620 스즈나안_t.ERB


code:
;-------------------------------------------------
;낚시를 한다
;외출 커맨드
;-------------------------------------------------
@COM621
#DIM 잡힌것
#DIM 낚시파워
#DIM 낚시파워M
#DIM 낚시파워T
#DIM 잡히지않는보정
#DIM 낚시운
#DIM CHARA
#DIM 상대자가낚시했다
#DIM 活性
#DIM FISHED, 2

상대자가낚시했다 = 0
;참가자 판정
IF FLAG:데이트상대
	CHARA = FLAG:데이트상대
ELSEIF TARGET && ABL:TARGET:친밀 >= 5 - TALENT:TARGET:낚시Lv
	CHARA = TARGET
ELSE
	CHARA = 0
ENDIF
;불쌍해；；
IF CHARA == [[와카사기히메]] || TARGET == [[와카사기히메]]
	PRINTFORMW 그녀 앞에서 낚시를 하는 것은 가혹한 것이다…
	RETURN -1
ENDIF

;季節ごとの活性
SELECTCASE DAY:2
	CASE 1  ;春
		活性 = 10
	CASE 2  ;夏
		活性 = 7
	CASE 3  ;秋
		活性 = 10
	CASE 4  ;冬
		活性 = 3
ENDSELECT
;時刻ごとの活性補正
SELECTCASE TIME:2
	CASE 1, 4  ;早朝と昼下がり
		活性 += 10
	CASE 5, 6  ;夕と夜
		活性 += 5
ENDSELECT
;天気による活性
SELECTCASE TIME:5
	CASE 3, 6 ;曇、霧雨
		活性 += 5
	CASE 4, 5 ;雨
		活性 += 7
	CASE 8 TO 13 ;雪系統はマイナス
		活性 -= 3
ENDSELECT
;虹が出てるときは、雨の後で水が濁っているだろう
SELECTCASE TIME:7
	CASE 1
		活性 += 7
ENDSELECT

;일일로―50~+50, 아이템이나 기원등으로 보정을 걸치는 일도 가능
낚시운 = FLAG:일일이벤트 / 10 - 50

;釣り技能で活性を察知
IF TALENT:MASTER:낚시Lv >= 3
	SETCOLOR C_L_GRAY
	SELECTCASE (活性 + 낚시운)
		CASE IS <= 10
			PRINTFORML （魚影がほとんど見えない…）
		CASE 11 TO 30
			PRINTFORML （魚影はまばらだ…）
		CASE 31 TO 50
			PRINTFORML （魚影はそれなりに見える…）
		CASE 51 TO 65
			PRINTFORML （魚影はかなり多く見える…）
		CASEELSE
			PRINTFORML （魚影がひしめき、水面がざわついている…）
	ENDSELECT
	RESETCOLOR
ENDIF

;낚시의 묘미 (참다운 즐거움)은 대기 시간이라고 들었다
FOR LOCAL, 0, 5
	REUSELASTLINE  \@ CHARA ?  %조사처리(CALLNAME:CHARA,"과")% 함께 #  \@낚시중. %". " * LOCAL%
	TWAIT 150, 1
NEXT
;낚시파워설정
낚시파워M = 10 + TALENT:MASTER:낚시Lv * 15 + 낚시운 + 잡히지않는보정
SIF CHARA
	낚시파워T = 10 + TALENT:CHARA:낚시Lv * 15 + 잡히지않는보정

;(낚시파워)%로 잡힌다
FISHED:0 = RAND:100
IF FISHED < 낚시파워M
	낚시파워= 낚시파워M
	$FISH
	;잡히면(자) 보정 리셋
	잡히지않는보정 = 0
	;낚시파워로 잡히는 것이 바뀐다
	SELECTCASE MIN(RAND:낚시파워, 100)
	CASE IS < 5
		;목재
		잡힌것 = 610
	CASE IS < 10
		;모리야부적
		잡힌것 = 131
	CASE 51 TO 60
		;개구리
		잡힌것 = 706
	CASE 10 TO 50, 61 TO 89
		FISHED:1 = (FISHED:0 + 10) / 10 * 10
		잡힌것 = FISHING_CATCH(FISHED:1)
	CASE IS < 95
		잡힌것 = FISHING_CATCH(95)
	CASEELSE
		잡힌것 = FISHING_CATCH(100)
		;보너스
		IF 상대자가낚시했다
			EXP:CHARA:낚시경험 ++
		ELSE
			EXP:MASTER:낚시경험 ++
		ENDIF
	ENDSELECT
	;아이템에 추가
	SELECTCASE 잡힌것
		CASE 1 TO 10, 22 TO 23
			CALL GET_ITEM("물고기",1,1)
		CASE 11 TO 20, 24 TO 60
			CALL GET_ITEM("맛있는물고기",1,1)
		CASE 21
			CALL GET_ITEM("로터",1,1)
		CASEELSE
			SIF ITEM:잡힌것 < 99
			ITEM:잡힌것 ++
	ENDSELECT
;CHARA가 낚시했을 경우
ELSEIF CHARA && RAND:100 < 낚시파워T
	낚시파워= 낚시파워T
	상대자가낚시했다 = 1
	GOTO FISH
;잡히지 않았으면 보정 가산, 잡힐 때까지 축적해 나간다
ELSE
	잡히지않는보정 += 10
	잡힌것 = 0
ENDIF

;기본의 경험
EXP:MASTER:낚시경험 ++

SIF ITEM:107 && !RAND:2
	EXP:MASTER:낚시경험 ++

SIF CHARA
	EXP:CHARA:낚시경험 ++
;낚시한 사람의 추가 경험
IF 잡힌것
	IF 상대자가낚시했다
		EXP:CHARA:낚시경험 ++
	ELSE
		EXP:MASTER:낚시경험 ++
	ENDIF
ENDIF
;데이트경험
IF CHK_DATENOW(CFLAG:CHARA:데이트중) && CHARA > 0
	CALL 붉은실발동
	EXP:MASTER:데이트경험 ++
	EXP:CHARA:데이트경험 ++
ENDIF

;당신이 낚시하면(자) 체력 감소
SIF 잡힌것 && !상대자가낚시했다
	DOWNBASE:MASTER:체력 = 50
DOWNBASE:MASTER:기력 = 100

;참가자의 소스, 적당하게 설정하고 있습니다, 동일 커맨드 제한을 받지 않기 때문에 조심스럽게, 다만 잡히지 않으면 매회 반감
IF CHARA > 0
	;고정으로 획득하는 소스
	SOURCE:CHARA:환락 = 400
	SOURCE:CHARA:사랑 = 50

	;ABL:순종을 본다
	IF ABL:순종 <= 1
		SOURCE:CHARA:환락 += (ABL:순종 * 40)
		SOURCE:CHARA:사랑 += (ABL:친밀 * 20)
	ELSEIF ABL:순종 <= 3
		SOURCE:CHARA:환락 += 200 + (ABL:순종 * 40)
		SOURCE:CHARA:사랑 += 50 + (ABL:친밀 * 30)
	ELSEIF ABL:순종 <= 5
		SOURCE:CHARA:환락 += 500 + (ABL:순종 * 40)
		SOURCE:CHARA:사랑 += 200 + (ABL:친밀 * 30)
	ELSEIF ABL:순종 <= 8
		SOURCE:CHARA:환락 += 750 + (ABL:순종 * 60)
		SOURCE:CHARA:사랑 += 300 + (ABL:친밀 * 40)
	ELSEIF ABL:순종 <= 11
		SOURCE:CHARA:환락 += 1000 + (ABL:순종 * 60)
		SOURCE:CHARA:사랑 += 600 + (ABL:친밀 * 40)
	ELSE
		SOURCE:CHARA:환락 += 1600 + (ABL:순종 * 30)
		SOURCE:CHARA:사랑 += 1000 + (ABL:친밀 * 20)
	ENDIF
	SIF CHARA != FLAG:데이트상대
		SOURCE:CHARA:사랑 = 0

	;호감 도는 보지 않는다

	SOURCE:CHARA:수동 = 200 + 100 * ABL:순종
	SOURCE:CHARA:정복 = 200 + 100 * ABL:새드끼

	;낚시질의 성과를 본다
	SELECTCASE 잡힌것
	;잡히지 않는다
	CASE 0
		TIMES SOURCE:CHARA:환락 , 0.50
		TIMES SOURCE:CHARA:수동 , 0.50
		TIMES SOURCE:CHARA:정복 , 0.50
	;목재, 모리야부적
	CASE 610, 131
		TIMES SOURCE:CHARA:환락 , 0.10
		TIMES SOURCE:CHARA:수동 , 0.50
		TIMES SOURCE:CHARA:정복 , 0.50
		SOURCE:CHARA:달성 = 20
	;황어, 아마고, 송어 , 이와나
	CASE 1 TO 10, 22
		TIMES SOURCE:CHARA:환락 , 1.00
		TIMES SOURCE:CHARA:수동 , 1.00
		TIMES SOURCE:CHARA:정복 , 1.00
		SOURCE:CHARA:달성 = 100
	;개구리
	CASE 706
		IF CHARA == [[치르노]]
			TIMES SOURCE:CHARA:환락 , 2.00
			TIMES SOURCE:CHARA:수동 , 2.00
			TIMES SOURCE:CHARA:정복 , 2.00
		ELSEIF CHARA == [[스와코]]
			TIMES SOURCE:CHARA:환락 , 0.50
			TIMES SOURCE:CHARA:수동 , 0.50
			TIMES SOURCE:CHARA:정복 , 0.50
		ELSE
			TIMES SOURCE:CHARA:환락 , 1.25
			TIMES SOURCE:CHARA:수동 , 1.00
			TIMES SOURCE:CHARA:정복 , 1.00
		ENDIF
		SOURCE:CHARA:달성 = 150
	;술, 빙어, 니지마스, 캄르치
	CASE 11 TO 13, 15 TO 20, 24 TO 30
		TIMES SOURCE:CHARA:환락 , 1.50
		TIMES SOURCE:CHARA:수동 , 1.50
		TIMES SOURCE:CHARA:정복 , 1.50
		SOURCE:CHARA:달성 = 200
	;칠성장어
	CASE 14
		;みすちー
		IF CHARA == 28
			TIMES SOURCE:CHARA:환락 , 2.00
			TIMES SOURCE:CHARA:수동 , 2.00
			TIMES SOURCE:CHARA:정복 , 2.00
		ELSE
			TIMES SOURCE:CHARA:환락 , 1.50
			TIMES SOURCE:CHARA:수동 , 1.50
			TIMES SOURCE:CHARA:정복 , 1.50
		ENDIF
		SOURCE:CHARA:달성 = 250
	;로터
	CASE 21
		SOURCE:CHARA:욕정 = 200
	;실러캔스，왕연어，얼음의비늘, 용의손톱，龍涎香
	CASE 23, 31, 609, 606, 643
		TIMES SOURCE:CHARA:환락 , 2.00
		TIMES SOURCE:CHARA:수동 , 2.00
		TIMES SOURCE:CHARA:정복 , 2.00
		SOURCE:CHARA:달성 = 300
	ENDSELECT
ENDIF

;시간경과
TIME += 20

;플래그 인도
TFLAG:SELECTCOM분기 = 잡힌것
TFLAG:SOURCE등에영향을주지않는SELECTCOM분기 = CHARA + (상대자가낚시했다 * 1000)

RETURN 1

;-------------------------------------------------
;실행 가능 판정
;-------------------------------------------------
;낚시를 한다
@COM_ABLE621
;일괄관리
SIF GLOBAL_COMABLE(621)
	RETURN RESULT
;낚싯대가 없다
SIF !ITEM:낚싯대
	RETURN 0
;거점에서는 할 수 없다(가짜)
SIF !FISHING_SPOT(CFLAG:MASTER:현재위치)
	RETURN 0
;데이트중은 데이트상대와 만
SIF FLAG:데이트상대 && TARGET != FLAG:데이트상대
	RETURN 0
;수면중
SIF CFLAG:MASTER:수면
	RETURN 0
;시간정지중
SIF FLAG:70
	RETURN 0
RETURN 1

;FileName_DATE_CMN.ERB ------------------------------- Rev1. 00
;낚시가 생기는 장소
;CALL		USER
;ARG		현재위치
;RETURN		할 수 있다면 1
;COMMENT	장소를 추가했을 경우, CASE를 변경의 일 현재 KR판에만 있음
;-----------------------------------------------------------
@FISHING_SPOT(ARG)
#FUNCTION
IF !IN_HOME(ARG)
	SELECTCASE ARG
		CASE 안개의호수
			RETURNF 1
		CASE 삼도천
			RETURNF 2
		CASE 현무의늪
			RETURNF 3
		CASE 큰두꺼비의연못
			RETURNF 4
		CASE 구천의폭포
			RETURNF 5
		CASE 산의호수
			RETURNF 6
		CASE 지옥의깊은길
			RETURNF 7
	ENDSELECT
ELSE
	SELECTCASE ARG
		CASE 347;霧の湖
			RETURNF 1
		CASE 602;賽の河原
			RETURNF 2
		CASE 703;玄武の沢
			RETURNF 3
		CASE 708;大蝦蟇の池
			RETURNF 4
		CASE 806;滝の裏
			RETURNF 5
		CASE 814;湖畔（山の湖）
			RETURNF 6
		CASE 945;川岸
			RETURNF 7
	ENDSELECT
ENDIF
RETURNF 0

;-------------------------------------------------
;釣れたもの判定の拡張　場所、季節、時刻に応じて可変
;ARG　…ランクに対応
;-------------------------------------------------
@FISHING_CATCH(ARG)
#FUNCTION

SELECTCASE ARG
	CASE 20
		SELECTCASE FISHING_SPOT(CFLAG:MASTER:현재위치)
			CASE 1, 4 ;버들치
				RETURNF 5
			CASE 2, 7 ;피라미
				RETURNF 6
			CASE 3, 5, 6 ;황어
				RETURNF 1
		ENDSELECT
	CASE 30
		SELECTCASE FISHING_SPOT(CFLAG:MASTER:현재위치)
			CASE 1, 2, 6 ;까나리
				RETURNF 7
			CASE 3, 5 ;송어
				RETURNF 2
			CASE 4 ;개구리
				RETURNF 706
			CASE 7 ;피라니아
				RETURNF 8
		ENDSELECT
	CASE 40
		SELECTCASE FISHING_SPOT(CFLAG:MASTER:현재위치)
			CASE 1, 2, 6 ;둑중개
				RETURNF 16
			CASE 3, 5 ;산천어ちゃん
				RETURNF 3
			CASE 4 ;개구리
				RETURNF 706
			CASE 7 ;가아
				RETURNF 9
		ENDSELECT
	CASE 50
		SELECTCASE FISHING_SPOT(CFLAG:MASTER:현재위치)
			CASE 1, 6 ;블루길
				RETURNF 10
			CASE 2, 7 ;은어（春夏）or피라미（秋冬）
				IF DAY:2 < 3
					RETURNF 17
				ELSE
					RETURNF 6
				ENDIF
			CASE 3, 5 ;곤들매기
				RETURNF 4
			CASE 4 ;개구리
				RETURNF 706
		ENDSELECT
	CASE 70
		SELECTCASE FISHING_SPOT(CFLAG:MASTER:현재위치)
			CASE 1, 4, 6 ;메기or빙어（冬）
				IF DAY:2 == 4
					RETURNF 12
				ELSE
					RETURNF 18
				ENDIF
			CASE 2, 7 ;뱀장어
				RETURNF 19
			CASE 3, 5 ;옥새송어or연어（秋）or빙어（冬）
				IF DAY:2 == 3
					RETURNF 11
				ELSEIF DAY:2 == 4
					RETURNF 12
				ELSE
					RETURNF 13
				ENDIF
		ENDSELECT
	CASE 80
		SELECTCASE FISHING_SPOT(CFLAG:MASTER:현재위치)
			CASE 1 ;각시송어
				RETURNF 20
			CASE 2, 4, 6 ;잉어
				RETURNF 22
			CASE 3, 5 ;칠성장어
				RETURNF 14
			CASE 7 ;철갑상어
				RETURNF 24
		ENDSELECT
	CASE 90
		SELECTCASE FISHING_SPOT(CFLAG:MASTER:현재위치)
			CASE 1 ;쿠니마스
				RETURNF 25
			CASE 2 ;초어
				RETURNF 26
			CASE 3 ;가물치
				RETURNF 15
			CASE 4 ;떡붕어
				RETURNF 27
			CASE 5 ;일본연어
				RETURNF 28
			CASE 6 ;큰입배스
				RETURNF 29
			CASE 7 ;피라루쿠
				RETURNF 30
		ENDSELECT
	CASE 95
		SELECTCASE  FISHING_SPOT(CFLAG:MASTER:현재위치)
			CASE 1 ;왕연어（新月の日）or로터
				IF MOON_WATCH(DAY:3) == 1
					RETURNF 31
				ELSE
					RETURNF 21
				ENDIF
			CASE 2 ;실러캔스
				RETURNF 23
			CASEELSE
				;로터
				RETURNF 21
		ENDSELECT
	CASE 100
		SELECTCASE FISHING_SPOT(CFLAG:MASTER:현재위치)
			CASE 1, 6 ;龍涎香or얼음의비늘（冬）
				IF DAY:2 == 4
					RETURNF 609
				ELSE
					RETURNF 643
				ENDIF
			CASE 2, 3, 5, 7 ;용의손톱or얼음의비늘（冬）
				IF DAY:2 == 4
					RETURNF 609
				ELSE
					RETURNF 606
				ENDIF
			CASE 4 ;개구리
				RETURNF 706
		ENDSELECT
ENDSELECT
RETURNF 0

;-----------------------------------------------------------
;釣った魚のナンバーを名称に変換
;コレクションリストにも利用するため、로터は除外しておく
;ARG　…釣れたものナンバーに対応
;-----------------------------------------------------------
@FISH_NAME(ARG)
#FUNCTIONS

SELECTCASE ARG
	CASE 1
		RETURNF "황어"
	CASE 2
		RETURNF "송어"
	CASE 3
		RETURNF "산천어"
	CASE 4
		RETURNF "곤들매기"
	CASE 5
		RETURNF "버들치"
	CASE 6
		RETURNF "피라미"
	CASE 7
		RETURNF "까나리"
	CASE 8
		RETURNF "피라니아"
	CASE 9
		RETURNF "가아"
	CASE 10
		RETURNF "블루길"
	CASE 11
		RETURNF "연어"
	CASE 12
		RETURNF "빙어"
	CASE 13
		RETURNF "옥새송어"
	CASE 14
		RETURNF "칠성장어"
	CASE 15
		RETURNF "가물치"
	CASE 16
		RETURNF "둑중개"
	CASE 17
		RETURNF "은어"
	CASE 18
		RETURNF "메기"
	CASE 19
		RETURNF "뱀장어"
	CASE 20
		RETURNF "각시송어"
	CASE 22
		RETURNF "잉어"
	CASE 23
		RETURNF "실러캔스"
	CASE 24
		RETURNF "철갑상어"
	CASE 25
		RETURNF "쿠니마스"
	CASE 26
		RETURNF "초어"
	CASE 27
		RETURNF "떡붕어"
	CASE 28
		RETURNF "일본연어"
	CASE 29
		RETURNF "큰입배스"
	CASE 30
		RETURNF "피라루쿠"
	CASE 31
		RETURNF "왕연어"
	CASEELSE
		RETURNF ""
ENDSELECT

;-------------------------------------------------
;釣れた魚のサイズを一般的サイズ内でランダムに出力する変数
;ARG　…釣れたものナンバーに対応
;-------------------------------------------------
@FISHING_SIZE(ARG)
#FUNCTION

LOCAL = 0
SELECTCASE ARG
	CASE 1 ;황어
		LOCAL = MAX(RAND:(56), 25)
	CASE 2 ;송어
		LOCAL = MAX(RAND:(40), 18)
	CASE 3 ;산천어ちゃん
		LOCAL = MAX(RAND:(50), 25)
	CASE 4 ;곤들매기
		LOCAL = MAX(RAND:(70), 20)
	CASE 5 ;버들치
		LOCAL = MAX(RAND:(18), 10)
	CASE 6 ;피라미
		LOCAL = MAX(RAND:(18), 10)
	CASE 7 ;까나리
		LOCAL = MAX(RAND:(36), 15)
	CASE 8 ;피라니아
		LOCAL = MAX(RAND:(61), 15)
	CASE 9 ;가아
		LOCAL = MAX(RAND:(300), 120)
	CASE 10 ;블루길
		LOCAL = MAX(RAND:(28), 12)
	CASE 11 ;연어
		LOCAL = MAX(RAND:(100), 65)
	CASE 12 ;빙어
		LOCAL = MAX(RAND:(18), 8)
	CASE 13 ;옥새송어
		LOCAL = MAX(RAND:(105), 35)
	CASE 14 ;칠성장어
		LOCAL = MAX(RAND:(65), 45)
	CASE 15 ;가물치
		LOCAL = MAX(RAND:(115), 80)
	CASE 16 ;둑중개
		LOCAL = MAX(RAND:(18), 10)
	CASE 17 ;은어
		LOCAL = MAX(RAND:(35), 25)
	CASE 18 ;메기
		LOCAL = MAX(RAND:(80), 50)
	CASE 19 ;뱀장어
		LOCAL = MAX(RAND:(120), 85)
	CASE 20 ;각시송어
		LOCAL = MAX(RAND:(55), 20)
	CASE 21 ;로터
		LOCAL = 5
	CASE 22 ;잉어
		LOCAL = MAX(RAND:(115), 55)
	CASE 23 ;실러캔스
		LOCAL = MAX(RAND:(200), 100)
	CASE 24 ;철갑상어
		LOCAL = MAX(RAND:(301), 100)
	CASE 25 ;쿠니마스
		LOCAL = MAX(RAND:(45), 30)
	CASE 26 ;초어
		LOCAL = MAX(RAND:(140), 70)
	CASE 27 ;떡붕어
		LOCAL = MAX(RAND:(62), 25)
	CASE 28 ;일본연어
		LOCAL = MAX(RAND:(220), 90)
	CASE 29 ;큰입배스
		LOCAL = MAX(RAND:(55), 25)
	CASE 30 ;피라루쿠
		LOCAL = MAX(RAND:(400), 250)
	CASE 31; 왕연어
		LOCAL = MAX(RAND:(550), 300)
ENDSELECT
RETURNF LOCAL

;-------------------------------------------------
;釣果の説明、消去行数の関係で５行に統一
;ARG　…釣れたものナンバーに対応
;-------------------------------------------------
@FISHING_EXPLANATION(ARG)
IF INRANGE(ARG, 11, 20) || INRANGE(ARG, 24, 60)
	CALL COLOR_PRINT("　　☆맛있는물고기☆", "C_YELLOW")
ELSE
	PRINTFORM 　　◇普通の魚◇
ENDIF
PRINTL
SELECTCASE ARG
	CASE 1 ;황어
		PRINTFORML 　　体長30cm～50cm前後、雑食性。背側は焦茶色、春には黒と赤の婚姻色を表す
		PRINTFORML 　　２～４年で成魚となり、群れて生活する。小骨が多く、臭いあり
		PRINTFORML 　　河川の上流から下流、はたまた山上湖にも生息。水質汚染や酸性水にも強い
		PRINTFORML
		PRINTFORML
	CASE 2 ;송어
		PRINTFORML 　　体長20cm～30cm前後、小魚や昆虫食。산천어とよく似た姿だが、側面に散在する朱点で区別できる
		PRINTFORML 　　サツキマスの陸封型（海や湖に降りず、河川で一生を終える）で、年間を通して水温の低い場所を好む
		PRINTFORML 　　漢字は「雨子」「雨魚」などが当てられ、梅雨時～初夏など雨がちな時期によく釣れる
		PRINTFORML 　　また、「甘い（美味しいの意）魚」という意味の異称を持つほど、美味である
		PRINTFORML
	CASE 3 ;산천어ちゃん
		PRINTFORML 　　体長は～40cm前後、ミミズや昆虫食。背側は緑がかった黄褐色、若いうちは小判状の斑紋模様（パーマーク）を示す
		PRINTFORML 　　サクラマスの陸封型（海や湖に降りず、河川で一生を終える）で、河川上流の冷水域に生息
		PRINTFORML 　　寿命は３～４年。生活スタイルの似る곤들매기よりも下流に生息し「棲み分け」を行う
		PRINTFORML 　　非常に警戒心が強く釣るのは難しいとされ、「渓流の女王」の異名をとる
		PRINTFORML
	CASE 4 ;곤들매기
		PRINTFORML 　　体長20～60cm前後、貪欲な肉食性で、蛇すら食した事例も。背側は緑褐色や灰色で、地域によって様々
		PRINTFORML 　　河川最上流の冷水域に生息し、生活スタイルの似る산천어よりも上流に暮らし「棲み分け」を行う
		PRINTFORML 　　かつては他の연어の仲間と同様に海に降りる生活をしていたようだが、本種はとりわけ冷水を好む性質ゆえ、
		PRINTFORML 　　氷河期の終焉とともに陸封型の生活に固定され、地方地方の河川ごとに分断された結果、多様な亜種が生まれたようだ
		PRINTFORML
	CASE 5 ;버들치
		PRINTFORML 　　体長～15cm程度、雑食性。体色は黄褐色で鱗が小さく、ぬめりが強いため「アブラ」の名をとる
		PRINTFORML 　　河川の中流から上流域、山地の湖沼など、低水温を好む。群れて産卵をする習性をもつ
		PRINTFORML 　　多少の苦みがあるが、天ぷらやマリネ、南蛮漬けなどに
		PRINTFORML
		PRINTFORML
	CASE 6 ;피라미
		PRINTFORML 　　体長～15cm程度、背側は灰白色、繁殖期のオスには青緑とピンクの鮮やかな婚姻色が表れる
		PRINTFORML 　　河川下流から中流域、湖沼などに生息。水質汚染や河川改修による環境変化に強い
		PRINTFORML 　　草食寄りの雑食性。甘露煮や天ぷら、なれずしなどに
		PRINTFORML
		PRINTFORML
	CASE 7 ;까나리
		PRINTFORML 　　キンブナは体長15cm程度、ギンブナは30cm程度まで成長。キンブナは黄褐色の体色をしている
		PRINTFORML 　　雑食性で、河川下流域や湖沼などに生息する잉어の仲間
		PRINTFORML 　　マブナとも言われるが、これはゲンゴロウブナ（떡붕어）と他の까나리を区別するための総称
		PRINTFORML 　　까나리の仲間の分類は非常に難しく、いまだに定説はないとされている
		PRINTFORML
	CASE 8 ;피라니아
		PRINTFORML 　　アマゾン川などの熱帯地方に生息する、鋭い歯を持つ肉食魚の総称
		PRINTFORML 　　現地の言葉で「歯のある魚」の意味で、小型種では15cmほど、大型種では60cmほどまで成長する
		PRINTFORML 　　「人食い魚」は誇張されたイメージで、実態は臆病。群れて暮らすが、群れ全体が興奮状態に入ると集団で攻撃を開始する
		PRINTFORML 　　肉はタンパク質豊富で美味。地底の地熱で温かい川に適応しているのだろうか？
		PRINTFORML
	CASE 9 ;가아
		PRINTFORML 　　堅牢な硬鱗（ガノイン鱗）を持つ、魚類の中でも原始的なスタイルの魚。北アメリカ大陸の原産
		PRINTFORML 　　細長い胴体と細長く伸びるアゴを持ち、種によっては３ｍに迫る巨体にまで成長することがある
		PRINTFORML 　　近年日本の某河川でも排水（温水）によって定着しているのではないかと言われている
		PRINTFORML 　　地底の地熱で温かい川に適応しているようだが……誰が持ち込んだのだろう？
		PRINTFORML
	CASE 10 ;블루길
		PRINTFORML 　　北米大陸が原産の魚。体長～25cm程度。湖沼や流れの緩やかな河川に生息する
		PRINTFORML 　　パンフィッシュ（フライパンに収まりやすい魚）と俗称されるグループの魚だけあり、丸っこく扁平な形をしている
		PRINTFORML 　　非常に悪食で、何でも食べる貪欲さが移植先の水棲生物に大きな影響を与えてしまい、駆除の対象となっている
		PRINTFORML 　　幻想郷にも分布を広げているのだとすれば、大変なことになりそうだ
		PRINTFORML
	CASE 11 ;연어
		PRINTFORML 　　体長70cm程度まで成長。川でふ化した稚魚は3～4月に海に降り、非常に長い距離を回遊しながら成長する
		PRINTFORML 　　４～６年ほど海で過ごした後、秋ごろに生まれた川を遡上し、湧水のある河床に産卵、一生を終える
		PRINTFORML 　　遡上してくる産卵期の個体の食味は卵や白子の形成によって脂・旨味が落ちており、沖合で漁獲されたものに及ばない
		PRINTFORML 　　……海のないはずの幻想郷で、一体何処でこれほどの成熟を遂げてくるのだろうか？
		PRINTFORML
	CASE 12 ;빙어
		PRINTFORML 　　体長は成魚で15cm程度。湖沼やそれに注ぐ河川に生息。一部の個体は海に降りることもある
		PRINTFORML 　　冬に産卵を行う一年魚で、冬・氷上の釣りがイメージされるが、実際は一年を通して釣れる
		PRINTFORML 　　冷水を好むとともに、存外に塩分や水質汚染にも強く、大量産卵をすることで多くの仲間を増やす繁殖力を併せ持つ
		PRINTFORML 　　食用にも向いており、諏訪湖をはじめ多くの湖に移植された
		PRINTFORML
	CASE 13 ;옥새송어
		PRINTFORML 　　別名レインボートラウト。体長は一般には40cm程度だが、湖などに住むものは80cm～1ｍ近くまで成長する
		PRINTFORML 　　産卵しても１回では死なず、数年にわたって産卵を行うことができる
		PRINTFORML 　　寄生虫がいないと言われ、刺身でも食べられる珍しい淡水魚で、繁殖期のオスには虹色の婚姻色が発現する
		PRINTFORML 　　日本に移入されたのは明治時代。幻想郷が大結界で隔てられた後の事だが……どこから入ってきたのだろう？
		PRINTFORML
	CASE 14 ;칠성장어
		PRINTFORML 　　体長は50～60cm。無顎類（アゴを持たない魚類）と呼ばれる極めて原始的な魚類で「生きた化石」
		PRINTFORML 　　体側に7対のエラの穴があり、それが目のように見えることから「八つ目」の名を持つ
		PRINTFORML 　　清浄な冷水を好み、降海型と陸封型が存在する。なお、뱀장어とは完全に別種である
		PRINTFORML 　　食味は硬くクセはあるが栄養は豊富で、ビタミンＡの一種を多量に含み、夜盲症に絶大な効果があるという
		PRINTFORML
	CASE 15 ;가물치
		PRINTFORML 　　別名ライギョ。体長は１ｍ程度まで成長する。水流が緩やかで、水草などが多い環境を好む
		PRINTFORML 　　魚食性だが、時にはカメや水鳥のヒナをも捕食する悪食。水底に潜み、待ち伏せを行う
		PRINTFORML 　　白身で淡白な味わいで、揚げ物や鍋で食される。タイワンドジョウの別称の通り、ユーラシア大陸が原産の外来種
		PRINTFORML 　　「そちら」の方面からやってきた誰かが持ち込んだものだろうか？
		PRINTFORML
	CASE 16 ;둑중개
		PRINTFORML 　　カサゴの仲間の淡水魚で体長15cm程度まで成長。体色は褐色～淡褐色まで変化に富む
		PRINTFORML 　　綺麗な水の河川の瀬や砂利底に潜んでいることが多く、水生昆虫や小魚をエサとする
		PRINTFORML 　　見た目は悪いが、とても美味でよいダシが出る。唐揚げや鍋、甘露煮など、広く料理に用いられる
		PRINTFORML
		PRINTFORML
	CASE 17 ;은어
		PRINTFORML 　　体長～30cm程度、成魚になると石についた藻類を食べて成長する。１年で寿命を終える魚
		PRINTFORML 　　成魚は縄張りを持ち、侵入した은어を追い払う習性をもつ。これを利用したのが囮の은어を泳がせて釣る「友釣り」
		PRINTFORML 　　本来は川と海を回遊する魚であるが、琵琶湖流域では海の代わりに琵琶湖に降り、遡上して大きく成長するものもいるという
		PRINTFORML 　　海のない幻想郷では、湖での成熟を行っているのだろうか？
		PRINTFORML
	CASE 18 ;메기
		PRINTFORML 　　体長は60cm程度。河川の中～下流域や湖沼、水田や用水路にも生息。夜行性で、発達した口ヒゲで周囲を探る肉食魚
		PRINTFORML 　　古くから、カエルをつけた仕掛けで水面を叩いて誘う「ポカン釣り」という手法で釣り上げられていた
		PRINTFORML 　　なお、地震の予兆として메기が暴れる、という俗信があるが、科学的な裏付けはまだ取れていないとされる
		PRINTFORML 　　地下に大메기がおり、これが暴れると大地震が起こる、というのも説話として有名……え？　夢に出た？
		PRINTFORML
	CASE 19 ;뱀장어
		PRINTFORML 　　体長は１ｍ程度まで成長する。河川の中～下流域や湖沼に生息。体表はヌルヌルした粘液に覆われる
		PRINTFORML 　　夜行性の肉食魚で、淡水魚に分類されるが、海に降って産卵・ふ化を行う降河回遊を行う
		PRINTFORML 　　深海でふ化し、海で成長してシラス뱀장어に、その後沿岸から川を遡上して成長を続け、５～１０年で成熟する
		PRINTFORML 　　その姿形から龍の化身、菩薩の遣いになぞらえられることもあり、食すのを禁忌とする地域もあったようだ
		PRINTFORML 　　めっきり外の世界で見なくなったのは、뱀장어が幻想入りしているから、なのだろうか？
	CASE 20 ;각시송어
		PRINTFORML 　　体長は20～50cm。北米大陸に生息するベニザケの陸封型（海や湖に降りず、河川で一生を終える魚）
		PRINTFORML 　　冷水を好むが、水深が深く冷たい湧水がある湖では暖地でも定着するため、いくつかの湖に移植された
		PRINTFORML 　　体色は灰色で、甲殻類などの動物プランクトンを主食とし、빙어とはエサを巡り競合を起こす
		PRINTFORML 　　연어科の魚の中でも最も美味とされるが、すぐに食味が落ちてしまう
		PRINTFORML
	;CASE 21 ;로터

	CASE 22 ;잉어
		PRINTFORML 　　体長は60cm～１ｍ程度まで成長。規模の大きい河川の下流～中流域、湖沼、用水路などに広く生息
		PRINTFORML 　　低水温や汚染に耐える生命力を持つうえ寿命が長く、70～80年も生き延びる個体も存在する
		PRINTFORML 　　水底を吸い混ぜ、およそ口に入るものなら何でも食べる強い雑食性を示す
		PRINTFORML 　　よく見られる体高の高い잉어は、中国から大昔に移入されたものであるといわれている
		PRINTFORML 　　琵琶湖水系などに生息するスリムな잉어が日本に古来から生息するものと推測され、野鯉と俗称される
	CASE 23 ;실러캔스
		PRINTFORML 　　体長は１ｍを超える。かつて広く世界中に分布した古代魚で、肉質のヒレを数多く持つ
		PRINTFORML 　　現生の陸上脊椎動物は全てこの系統から進化したと言われており、深海から生存種が発見された生きた化石
		PRINTFORML 　　味は申し訳ないがまずい。カニの肉をさらに薄味にしたような味、ハブラシのような食味、と評されている
		PRINTFORML 　　この個体は淡水性で、外の世界ではとうに絶滅したはずの種と考えられる……幽霊？
		PRINTFORML
	CASE 24 ;철갑상어
		PRINTFORML 　　体長１ｍ～４ｍにも成長。形はサメに似るが、サメとは別種の古代魚
		PRINTFORML 　　卵を塩漬けにしたものがキャビアであり、各地で乱獲の憂き目にあった
		PRINTFORML 　　かつて日本にも生息していたが、２０世紀初頭に絶滅したという
		PRINTFORML 　　現代は温泉水の熱などを利用し、外国産の品種の養殖も行われている
		PRINTFORML
	CASE 25 ;쿠니마스
		PRINTFORML 　　体長30cm～40cm。体色は灰色から黒色。각시송어の亜種とも言われたが別種の可能性もあるという
		PRINTFORML 　　水深の深い湖に生息し、産卵時は30～40ｍの比較的浅い所まで上がってくるという
		PRINTFORML 　　もとは秋田県の田沢湖にのみ生息した固有種であったが、湖に酸性の河川水を引き入れた事により絶滅
		PRINTFORML 　　しかし、昭和初期に山梨県の西湖に卵が移入されており、その子孫が絶滅宣言から70年後に再発見された
		PRINTFORML 　　どう食べても非常に美味、と評されており、かつては特産品として献上されていた
	CASE 26 ;초어
		PRINTFORML 　　体長は1.2ｍ程度、高水温や酸欠にも強い、水草や水際の雑草を食する草食性の잉어の仲間
		PRINTFORML 　　もとは中国の大河に生息しており、上流で産卵した浮遊性の卵が水流に流されるうちにふ化する特性をもつ
		PRINTFORML 　　卵が海まで流されると死んでしまうため、流れの緩やかで長大な流域面積を持つ河でないと繁殖できない
		PRINTFORML 　　中国では四大家魚と称されるほどポピュラーな魚で、蒸し魚・煮魚・揚げ物・スープなど、広く食べられている
		PRINTFORML
	CASE 27 ;떡붕어
		PRINTFORML 　　琵琶湖水系のゲンゴロウブナの変種を品種改良したもので、横から見るとひし形の形をしている
		PRINTFORML 　　体長は30～60cm程度。寿命は長く、数十年生きるものもいるという
		PRINTFORML 　　「釣りは까나리に始まり까나리に終わる」の金言の、「終わりの까나리」が本種であるとされ、難易度が高い
		PRINTFORML 　　季節により別種であるように行動が変わり、水温や天気によって遊泳層（タナ）も変わるうえ、
		PRINTFORML 　　採餌も溶け出したエサを小さく吸い込むだけで、感度の高いウキや竿を用いなければアタリをとるのも難しい
	CASE 28 ;일본연어
		PRINTFORML 　　体長1～1.5ｍ、時に２ｍを超える日本最大の淡水魚。冷水で流れの緩い河川の中流～下流域に生息
		PRINTFORML 　　背は青みがかった褐色。かつては本州北部にも生息していたが、現在は北海道にのみ分布する
		PRINTFORML 　　日本の일본연어は연어のように降海するが、幻想郷の本種は外国種のように陸封型の生活を送っているようだ
		PRINTFORML 　　肉は白身で脂が乗って美味しいが、成長が遅いのが難点
		PRINTFORML
	CASE 29 ;큰입배스
		PRINTFORML 　　体長30～50cm、北米原産の肉食魚で、湖沼や河川の中～下流域に生息
		PRINTFORML 　　自分の体長の半分ほどの魚や、小鳥やカエル、ネズミまで丸呑みにしてみせる貪欲さを持つ
		PRINTFORML 　　スポーツフィッシングの対象として有名だが、移入先で生態系を圧迫するなど問題も多い
		PRINTFORML 　　肉は淡白な白身でクセが無く、ムニエルやフライなどでおいしく頂ける
		PRINTFORML
	CASE 30 ;피라루쿠
		PRINTFORML 　　南米大陸に生息している体長３ｍを超える大魚。魚食性で、丸太のような胴、平たい頭部と尾ひれが特徴
		PRINTFORML 　　世界最大級の淡水魚の一種で、ガーの仲間と並んで古くから姿を変えていない古代魚
		PRINTFORML 　　体色は黒から銀色まで様々で、成魚では後部が赤色に変色する。現地の言葉で「紅い魚」と呼ばれる由縁
		PRINTFORML 　　肉は食べごたえがあり、無数の突起状の骨の通った舌（骨舌）はおろし器、鱗はヤスリや靴べらに、と広く利用された
		PRINTFORML
	CASE 31; 왕연어
		PRINTFORML 　　体長３ｍ以上にもなるという大怪魚。大型のマスのような姿で口は兎のような三口、下あごが突き出ている
		PRINTFORML 　　冷たく深い湖に生息しているとされ、その実態はいまだ掴めていない謎の魚
		PRINTFORML 　　肉は赤身で多くの脂を含み、簡単には焼けないが非常に美味であるという
		PRINTFORML 　　正体は、全くの新種の魚か、곤들매기の変種か、湖に残留する일본연어か、はたまたキングサーモンか
		PRINTFORML 　　その真相は、今も深く冷たい水の中に隠れている……
ENDSELECT

path: ERB/커맨드관련/COMF/외출계/COMF621 낚시한다_t.ERB


code:
@COM622
PRINTL 소지 아이템
CALL LIST_ITEM, "야채"
PRINTL
PRINTL
PRINTFORML 구입할 아이템을 선택해주세요 소지금:\\{MONEY}

CALL SHOW_ITEM, "야채"
PRINT [100] - 돌아간다
$INPUT_LOOP
INPUT
IF RESULT == 100
	TIME += 10
	RETURN -1
ELSEIF ITEM_TYPES( RESULT, "야채" )
	CALL BUY_ABLE, RESULT
	PRINTL
	RESTART
ELSE
	GOTO INPUT_LOOP
ENDIF
TIME += 10

;-------------------------------------------------
;実行可能判定
;-------------------------------------------------
;八百屋
@COM_ABLE622
;一括管理
SIF GLOBAL_COMABLE(622)
	RETURN RESULT
;八百屋ではない場所
SIF !YASAI_SHOP(CFLAG:MASTER:현재위치)
	RETURN 0
;수면中
SIF CFLAG:수면
	RETURN 0
;시간정지中
SIF FLAG:70
	RETURN 0
RETURN 1

@YASAI_SHOP(ARG)
#FUNCTION
;八百屋
SIF ARG == GET_MAP_REPLACEMENT(227)
	RETURNF 1
RETURNF 0

path: ERB/커맨드관련/COMF/외출계/COMF622 야채가게.ERB


code:
;목재상
@COM623
#DIM 제재판정
#DIM 판매가
#DIM 루프수
#DIM 결과보관
#DIM 후보번호
#DIM 후보개수
#DIM 매각개수
#DIM ProcessingCharge
루프수 = 0

;開店時間・閉店時間
IF INRANGE(TIME:0, 540, 1080) == 0
	PRINTFORMW 영업시간 외 같다
	RETURN -1
ENDIF

;初回イベント
IF FLAG:마지막목재상방문일 == 0
	;初回イベント前は、통나무持ちでなければ店にも入れない
	IF ITEM:통나무 == 0
		PRINTFORMW 特に今は用はない…売り物になる材木を伐ってこなくては
		RETURN -1
	ENDIF
	;依頼の報酬など、伐採以外で통나무を入手してしまった場合の退避用
	SIF !FLAG:통나무종류
		FLAG:통나무종류 = MAX(RAND:12, 1)
	PRINTFORMW 「はいどうも、私がここの主です。お客さまが、통나무を納めたいという方で？」
	PRINTFORMW 「では、品のほうを拝見させていただきましょうか」
	PRINTFORMW 「………」
	PRINTFORMW 「ほう、なかなかのものですね……」
	PRINTFORMW 「……え？　その気になれば、毎日でも持って来られる？　里の外ならいっぱい生えてるから？」
	PRINTFORMW 「…………」
	PRINTFORMW 「……ただ者じゃあないね、お客さん」
	PRINTFORMW 「いいだろう。伐採してきた통나무、ウチで買い取らせてもらおう」
	PRINTFORMW 「もちろん、材として価値のあるものは高く、そうでないものは安く買い取るぞ」
	PRINTFORMW 「家具やらなにやらで제재が必要なら、持ってきた통나무を제재に加工してやってもいい。それなりに作業代はいただくがな」
	PRINTFORMW 「それじゃあ、今回の品物をもう一度検めさせてもらおうか」
	;판매가はここで設定、基本的に中途キャンセルはしない
	판매가 = WOOD_VALUE(FLAG:통나무종류)
	SETCOLOR 0x00FF00
	PRINTFORMD 「
	PRINTFORM %WOOD_NAME(FLAG:통나무종류)%
	PRINTFORMD だな……
	PRINTFORM \\{판매가}
	RESETCOLOR
	PRINTFORMW で買い取らせてもらおう」
	PRINTFORMW 　　所持金: \\%TOSTR(MONEY,"#,###")% → \\%TOSTR(MONEY + 판매가,"#,###")%
	PRINTFORMW 「これで取引終了だ。じゃあ、次の商品も待ってるぞ」
	PRINTFORMW 「なに、妖精や妖怪が跋扈する山林から木を伐り出してこれる木こりなんて、そう多くはないからな。期待してるぞ？」
	PRINTW
	MONEY += 판매가
	ITEM:통나무 = 0
	FLAG:통나무종류 = 0
	루프수 = 1
ENDIF

$LOOP_SELECT

IF 루프수
	PRINTL
	PRINTFORMW 「さて、ほかにも用はあるか？」
ELSE
	;平常来店
	PRINTL
	IF DAY:0 - FLAG:마지막목재상방문일 > 28
		PRINTFORMW 「おう、なんだ。てっきり山仕事からは足を洗ったのかと思ったぞ」
	ELSEIF DAY:0 - FLAG:마지막목재상방문일 > 14
		PRINTFORMW 「おう、久しぶりだな」
	ELSEIF FLAG:마지막목재상방문일 == DAY:0
		PRINTFORMW 「なんだ、忘れ物か？　せわしないやつだな」
	ELSE
		PRINTFORMW 「おう、今日は何の用だ？」
	ENDIF
ENDIF

루프수 ++
FLAG:마지막목재상방문일 = DAY:0

PRINTFORML 　　소지금: \\%TOSTR(MONEY,"#,###")%
DRAWLINE

CALL ASK_M("통나무매각・제재", ITEM:통나무, "목재를 판다", ITEM:목재, "제재를 판다", 4, "取引内容を確認する", 5, "店を出る", 6)
결과보관 = RESULT
ProcessingCharge = LUMBERING(FLAG:통나무종류, "목재상")
CLEARLINE 12
SELECTCASE 결과보관
	;통나무を売る　제재してほしい
	CASE 0
		;依頼の報酬など、伐採以外で통나무を入手してしまった場合の退避用
		SIF !FLAG:통나무종류
			FLAG:통나무종류 = MAX(RAND:12, 1)
		SETCOLOR 0x00FF00
		PRINTFORML 「%WOOD_NAME(FLAG:통나무종류)%だな……」
		판매가 = WOOD_VALUE(FLAG:통나무종류)
		PRINTFORMD 「買取なら買値は
		PRINTFORM \\%TOSTR(판매가,"#,###")%
		PRINTFORMD 、제재するなら手間賃
		PRINTFORM \\%TOSTR(ProcessingCharge,"#,###")%
		RESETCOLOR
		PRINTFORMW いただくぞ」
		CALL ASK_M("매각", 1, "제재", MONEY > ProcessingCharge,"やっぱり그만둔다", 1)
		SELECTCASE RESULT
			CASE 0
				PRINTFORMW 　所持金: \\ → \\%TOSTR(MONEY + 판매가,"#,###")%
				PRINTFORMW 「毎度。じゃあ、次の商品を待ってるぞ」
				PRINTL
				MONEY += 판매가
			CASE 1
				PRINTL
				PRINTFORMW 　所持金: \\%TOSTR(MONEY,"#,###")% → \\%TOSTR(MONEY - ProcessingCharge,"#,###")%
				PRINT
				CALL PAY_MONEY_YEN(ProcessingCharge, 2)
				PRINTL
				PRINTFORMW 「はいよ。じゃあ、作業にかからせよう」
				CALL LogRefine("목재상")
			CASE 2
				PRINTFORMW 「なんだ、別にアテがあるのか？」
				PRINTL
				GOTO LOOP_SELECT
		ENDSELECT
		ITEM:통나무 = 0
		FLAG:통나무종류 = 0
	;목재を売る
	CASE 1
		후보개수 = ITEM:목재
		후보번호 = 610
		PRINTFORM 「%ITEMNAME:후보번호%なら　\\%TOSTR(ITEMPRICE:후보번호 / 2,"#,###")%で買い取るが、どうだ？」
		CALL COM623_SELL(후보개수)
		SELECTCASE RESULT
			CASE 0
				PRINTL
				GOTO LOOP_SELECT
			CASE 1 TO 후보개수
				CLEARLINE 2
				매각개수 = RESULT
				PRINTFORML 「それじゃあ%ITEMNAME:후보번호%を{매각개수}つ、\\%TOSTR((ITEMPRICE:(후보번호) / 2 * 매각개수),"#,###")%で買い取らせてもらおう」
				PRINTFORMW 　所持金: \\%TOSTR(MONEY,"#,###")% → \\%TOSTR(MONEY + (ITEMPRICE:(후보번호) / 2 * 매각개수),"#,###")%
				MONEY += (ITEMPRICE:(후보번호) / 2 * 매각개수)
				ITEM:후보번호 -= 매각개수
				PRINTL
				GOTO LOOP_SELECT
		ENDSELECT
	;제재を売る
	CASE 2
		제재판정 = 0
		FOR LOCAL, 1, 50
			SIF LOG_SALES:LOCAL
				BREAK
			제재판정 ++
		NEXT
		IF 제재판정 > 48
			PRINTFORMW 제재を持ってない
			PRINTL
			GOTO LOOP_SELECT
		ENDIF
		PRINTFORMW 「제재を仕入れてきたのか。どれどれ」
		DRAWLINE
		FOR LOCAL, 1, 50
			SIF LOG_SALES:LOCAL == 0
				CONTINUE
			PRINTFORM
			SIF LOCAL < 10
				PRINTFORM
			PRINTFORM [{LOCAL}]
			PRINTFORM %WOOD_NAME(LOCAL), 16, LEFT%
			PRINTFORM 제재数：
			SELECTCASE LOG_SALES:LOCAL
				CASE IS < 10
					PRINTFORM
				CASE IS < 100
					PRINTFORM
				CASE IS < 1000
					PRINTFORM
				CASE IS < 10000
					PRINTFORM
			ENDSELECT
			PRINTFORML {LOG_SALES:LOCAL}
		NEXT
		PRINTFORML 　 [0] 그만둔다

		$INPUT_LOOP03
		INPUT

		결과보관 = RESULT
		IF 결과보관 && LOG_SALES:결과보관
			판매가 = WOOD_VALUE(결과보관, "제재")
			후보개수 = LOG_SALES:결과보관
			SETCOLOR 0x00FF00
			PRINTFORMD 「
			PRINTFORM %WOOD_NAME(결과보관)%
			PRINTFORMD の제재なら
			PRINTFORM \\%TOSTR(판매가,"#,###")%
			RESETCOLOR
			PRINTFORML で買い取るが、どうだ？」
			CALL COM623_SELL(후보개수)
			SELECTCASE RESULT
				CASE 0
					PRINTL
					GOTO LOOP_SELECT
				CASE 1 TO 후보개수
					CLEARLINE 2
					매각개수 = RESULT
					PRINTFORML 「それじゃあ%WOOD_NAME(결과보관)%の제재を{매각개수}本、\\%TOSTR(판매가 * 매각개수,"#,###")%で買い取らせてもらおう」
					PRINTFORMW 　所持金: \\%TOSTR(MONEY,"#,###")% → \\%TOSTR(MONEY + (판매가 * 매각개수),"#,###")%
					MONEY += 판매가 * 매각개수
					LOG_SALES:결과보관 -= 매각개수
					PRINTL
			ENDSELECT
		ELSEIF 결과보관 == 0
			PRINTL
			GOTO LOOP_SELECT
		ELSE
			CLEARLINE 1
			GOTO INPUT_LOOP03
		ENDIF
	;取引内容を確認する
	CASE 3
		PRINTFORMW 「持ち込みの통나무や竹材、제재を買い取り中だ」
		PRINTFORMW 「持て余してるなら、端材・목재の類も引き取るぞ。まあ、二束三文にしかならないがな」
		PRINTFORMW 「必要なら통나무の제재も請け負おう。ただし、통나무の種類・大きさに応じて作業費を払ってもらう」
		PRINTFORMW 「ちなみに、此処で제재したものを此処に卸しても損しかしないからな。寄付のつもりなら歓迎するが」
		PRINTFORMW 「……なに、家具が欲しい？　それはうちの領分じゃないな。家具職人をあたってくれよ」
		PRINTL
	CASE 4
		PRINTFORMW 「おう、それじゃあな」
		PRINTL
		TIME += 루프수 * 5
		RETURN -1
ENDSELECT
GOTO LOOP_SELECT

;-------------------------------------------------
;実行可能判定
;-------------------------------------------------
;목재상
@COM_ABLE623
;一括管理
SIF GLOBAL_COMABLE(623)
	RETURN RESULT

IF !AT_HOME(MASTER)
	SIF CFLAG:MASTER:현재위치 != 상가거리
		RETURN 0
ELSE
	;西通り
	SIF CFLAG:MASTER:현재위치 != 206
		RETURN 0
ENDIF
;睡眠中
SIF CFLAG:수면
	RETURN 0
;時間停止中
SIF FLAG:70
	RETURN 0
RETURN 1

;-------------------------------------------------
;통나무・제재の매각価格を設定する関数
;何回か見る場合もあるので、その日の中では固定値出力
;ARG:0		FLAG:통나무종류
;Type		매각価格  省略時は통나무
;-------------------------------------------------
@WOOD_VALUE(ARG:0, Type)
#FUNCTION
#DIM Species
#DIM 판매가베이스
#DIM 입담보정
#DIMS Type

Species = ARG % 100

IF Type == "제재"
	판매가베이스 =WoodValue_Plank:Species
ELSE
	판매가베이스 = WoodValue_Log:Species
ENDIF

;日替わり変動
판매가베이스 += FLAG:일일이벤트 * 2

;입담보정  日付で乱数を固定
RANDOMIZE DAY:0
입담보정 = RAND:(1 + ABL:MASTER:화술기능 * 5 + ABL:MASTER:교양)

SELECTCASE 입담보정
	CASE 0 TO 10

	CASE 11 TO 20
		TIMES 판매가베이스, 1.05
	CASE 21 TO 30
		TIMES 판매가베이스, 1.07
	CASEELSE
		TIMES 판매가베이스, 1.10
ENDSELECT

;통나무の時のみ、巨木・大木補正
IF Type != "제재"
	IF ARG > 200
		판매가베이스 *= 5
	ELSEIF ARG > 100
		판매가베이스 *= 3
	ENDIF
ENDIF
RETURNF 판매가베이스

;-------------------------------------------------
;FLAG:통나무종류から、巨木・大木を無視して목재種番号を取り出す関数
;ARG…FLAG:통나무종류
;-------------------------------------------------
@WOOD_NUM(ARG)
#FUNCTION

RETURNF ARG % 100

;-------------------------------------------------
;제재コスト、제재数を割り出す関数
;複数回選べるので、固定値を出す
;ARG	FLAG:통나무종류
;Type	価格ないし제재数
;-------------------------------------------------
@LUMBERING(ARG, Craft, Type)
#FUNCTION
#DIM 비용베이스
#DIMS Craft
#DIMS Type

IF Type == "QTY"
	RETURNF 1 + (ARG / 100) * 2
ENDIF

비용베이스 = WOOD_VALUE(ARG)

SELECTCASE Craft
	CASE "목재상"
		RETURNF 비용베이스 / 2
	CASE "제재소"
		RETURNF 비용베이스 / 4
ENDSELECT

RETURNF 0

@COM623_SELL(후보개수)
#DIM 후보개수
SETCOLOR 0x777777
PRINTFORML 　몇 개 판매하겠습니까? (1～{후보개수}개 숫자를 입력、0으로 돌아간다)
RESETCOLOR
PRINTL
PRINT
PRINTBUTTON "[0] 그만둔다", 0
PRINT
PRINTBUTTON "[1] 1개", 1
PRINT
IF 후보개수 >= 5
	PRINTBUTTON "[5] 5개", 5
	PRINT
ENDIF
IF 후보개수 >= 10
	PRINTBUTTON "[10] 10개", 10
	PRINT
ENDIF
INPUT
SIF RESULT > 후보개수
	RETURN 0
SIF RESULT <= 0
	RETURN 0
RETURN RESULT

path: ERB/커맨드관련/COMF/외출계/COMF623 목재상.ERB


code:
;산의제재소
@COM624
#DIM 판매가
#DIM 루프수
#DIM 결과보관

루프수 = 0

;開店時間・閉店時間
IF INRANGE(TIME:0, 540, 1080) == 0
	PRINTFORMW 영업시간 외 같다
	RETURN -1
ENDIF

;初回イベント
IF FLAG:마지막제재소방문일 == 0
	;初回イベント前は、통나무持ちでなければ店にも入れない
	IF ITEM:통나무 == 0
		PRINTFORMW 特に今は用はない…売り物になる材木を伐ってこなくては
		RETURN -1
	ENDIF
	;依頼の報酬など、伐採以外で통나무を入手してしまった場合の退避用
	SIF !FLAG:통나무종류
		FLAG:통나무종류 = MAX(RAND:12, 1)

	PRINTFORMW 「はじめまして、盟友」
	SIF FLAG:역극 == 0
		PRINTFORMW 「噂は聞いてるよー。やたらアクティブで、妙な外来人がいるって」
	SIF ABL:[[니토리]]:친밀 >= 3
		PRINTFORMW 「캇파の仲間も世話になってるみたいだね。良いカ……げふんげふん、善良で信頼のおけるヤツだって」
	PRINTFORMW 「で、何の用？　へえ、통나무から製材してほしい、と。いいよー、工賃はこんな感じで――」
	PRINTFORMW ずずい、と캇파が説明用のパネルを突き出してくるが……
	PRINTFORMW これ、人里の相場と比べてだいぶ安くないか？
	PRINTFORMW 「そりゃそうさ。工作機械による全自動製材が、人力に後れをとるわけないでしょ」
	PRINTFORMW 「……とはいえ、寺の建立や神社の修復とかが終わって注文が少なくて、せっかくこさえた機械も持て余し気味でさ」
	PRINTFORMW 「そうだ。今後ともご贔屓にしてもらえるなら、今日は初回サービスってことで、タダでやってあげるけど？」
	PRINTFORMW …それ、自分の機械を自慢したいだけなんじゃないの、という言葉は飲み込んで、ご厚意に甘えることにした
	PRINTFORMW 「決まりだね。それじゃあ통나무をここにセットしてくれ」
	PRINTFORMW 「では、スイッチ・オーン！　ふふふ、唸れ我がマシン！　캇파の技術力の粋を見よ！」
	PRINTFORMW 妙にテンションが上がってる캇파はさておき、工作機械は緻密かつ高速に製材を仕上げていった
	PRINTFORMW …………
	PRINTFORMW ……
	CALL LogRefine("제재소")
	PRINTL
	PRINTFORMW 「はい完成！　どうよ！」
	PRINTFORMW ふんすー、とドヤ顔を浮かべる캇파に、素直に感謝を述べておいた
	PRINTFORMW 「へへ、お眼鏡に適ったのなら何よりだよ。じゃあ、今後ともご贔屓にねー」
	PRINTL
	루프수 = 1
ENDIF

$LOOP_SELECT

IF 루프수
	PRINTL
	PRINTFORMW 「さあさあ、次の통나무を持ってきてくれよ」
ELSE
	;平常来店
	PRINTL
	IF FLAG:마지막제재소방문일 == DAY:0
		PRINTFORMW 「およ？　どしたの？」
	ELSE
		PRINTFORM 「やあ、
		SELECTCASE TIME:0
			CASE 540 TO 600
				PRINTFORM おはよう
			CASE 601 TO 1020
				PRINTFORM こんにちわ
			CASEELSE
				PRINTFORM こんばんわ
		ENDSELECT
		PRINTFORM 盟友。
		IF DAY:0 - FLAG:마지막제재소방문일 > 28
			PRINTFORMW しばらく顔を見なかったから、心配してたんだ」
		ELSEIF DAY:0 - FLAG:마지막제재소방문일 > 14
			PRINTFORMW 久しぶりだね」
		ELSE
			PRINTFORMW 今日は何の用かな？」
		ENDIF
	ENDIF
ENDIF

루프수 ++
FLAG:마지막제재소방문일 = DAY:0

PRINTFORML 　　소지금: %TOSTR(MONEY,"#,###")%　소지카리스마: %TOSTR(MONEY:2,"#,###")%
DRAWLINE

CALL ASK_M("製材してほしい", ITEM:통나무, "取引内容を確認する", 1, "店を出る", 2)
CLEARLINE 7

SELECTCASE RESULT
	;製材してほしい
	CASE 0
		;依頼の報酬など、伐採以外で통나무を入手してしまった場合の退避用
		SIF !FLAG:통나무종류
			FLAG:통나무종류 = MAX(RAND:12, 1)

		SETCOLOR 0x00FF00
		PRINTFORMD 「
		PRINTFORM %WOOD_NAME(FLAG:통나무종류)%
		PRINTFORMD の製材だね。工賃は
		PRINTFORM \\%TOSTR(LUMBERING(FLAG:통나무종류, "제재소"),"#,###")%
		PRINTFORMD でやったげるよ。ああ、
		PRINTFORM %TOSTR(LUMBERING_CMCOST(FLAG:통나무종류),"#,###")%カリスマ
		RESETCOLOR
		PRINTFORMW でもいいよ？」
		CALL ASK_M("\\で支払う", MONEY >= LUMBERING(FLAG:통나무종류, "제재소"), "カリスマで支払う", MONEY:2 >= LUMBERING_CMCOST(FLAG:통나무종류), "やっぱりやめる", 2)

		결과보관 = RESULT
		IF 결과보관 == 0 || 결과보관 == 1
			IF 결과보관 == 0
				PRINTFORMW 　소지금: \\%TOSTR(MONEY,"#,###")% → \\%TOSTR(MONEY - LUMBERING(FLAG:통나무종류, "제재소"),"#,###")%
				PRINT
				CALL PAY_MONEY_YEN(LUMBERING(FLAG:통나무종류, "제재소"), 2)
			ELSE
				PRINTFORMW 　소지카리스마: %TOSTR(MONEY:2,"#,###")% → %TOSTR(MONEY:2 - LUMBERING_CMCOST(FLAG:통나무종류),"#,###")%
				PRINT
				CALL PAY_MONEY_CM(LUMBERING_CMCOST(FLAG:통나무종류), 2)
			ENDIF
			PRINTL
			PRINTDATAW
				DATAFORM 「まいどー。それじゃあ、スイッチ・オーン！　ふふふ、唸れ我がマシン！　캇파の技術力の粋を見よ！」
				DATAFORM 「まいど。ここに통나무をセットして、あとはフルオートでお任せだよ！」
				DATAFORM 「まいどー。では、ファクトリー・オートメーションの神髄をお見せしよう！」
			ENDDATA
			CALL LogRefine("제재소")
			PRINTL
		ELSEIF 결과보관 == 2
			PRINTFORMW 「む、値切り交渉は受け付けないからねー」
			PRINTL
			GOTO LOOP_SELECT
		ENDIF
	;取引内容を確認する
	CASE 1
		PRINTFORMW 「통나무の製材、格安で承ってるよー」
		PRINTFORMW 「통나무や製材の買い取り？　山ではその手の物資は供給過多だからね、人里で卸したほうがいい値がつくんじゃない？」
		PRINTFORMW 「家具とか作れないかって？　うーん。私ら飽きっぽいから、あんまりそういうのに向いてないんだよね」
		PRINTFORMW 「そういう手の込んだ仕事は、人間の職人たちに頼むのがいいと思うよ」
		PRINTFORMW 「私らが作ると、気の向くまま、いらない機能をゴテゴテ載せた캇파家具になっちゃいそうだしねー」
		PRINTFORMW 「…それはそれで興味がある？　へへ、まぁそのうちにねー」
		PRINTL
	CASE 2
		PRINTFORMW 「じゃあ、またね」
		PRINTL
		TIME += 루프수 * 5
		RETURN 1
ENDSELECT
GOTO LOOP_SELECT

;-------------------------------------------------
;実行可能判定
;-------------------------------------------------
;山の제재소
@COM_ABLE624
;一括管理
SIF GLOBAL_COMABLE(624)
	RETURN RESULT

IF !AT_HOME(MASTER)
	SIF CFLAG:MASTER:현재위치 != 현무의늪
		RETURN 0
ELSE
	;山麓が拠点の場合は、캇파のアジト
	SIF CFLAG:MASTER:현재위치 != 704
		RETURN 0
ENDIF
;睡眠中
SIF CFLAG:수면
	RETURN 0
;時間停止中
SIF FLAG:70
	RETURN 0
RETURN 1

;-------------------------------------------------
;製材コストをカリスマに変換する関数
;ARG …FLAG:통나무종류
;-------------------------------------------------
@LUMBERING_CMCOST(ARG)
#FUNCTION
#DIM 비용베이스

비용베이스 = LUMBERING(ARG, 1, 0)

;元値の幅がそんなに広くないので、一律で
RETURNF 비용베이스 / 300

;-------------------------------------------------
;製材結果の表示など
;-----------------------------------------------
@LogRefine(Craft)
#DIMS Craft
LOCAL = LUMBERING(FLAG:통나무종류, Craft, "QYT")
SETCOLOR 0x00FF00
PRINTFORMW 　%WOOD_NAME(FLAG:통나무종류, 1)%の製材: {LOG_SALES:WOOD_NUM(FLAG:통나무종류)} → {LOG_SALES:WOOD_NUM(FLAG:통나무종류) + LOCAL}
LOG_SALES:WOOD_NUM(FLAG:통나무종류) += LOCAL
ITEM:통나무 = 0
FLAG:통나무종류 = 0
RESETCOLOR

path: ERB/커맨드관련/COMF/외출계/COMF624 산의제재소.ERB


code:
;가구점
@COM625
#DIM 판매가
#DIM 루프수
#DIM 결과
#DIM 자재보관
#DIM 삭제행수
#DIM MaterialQTY
#DIMS DiscountedPrice

루프수 = 0

;開店時間・閉店時間
IF INRANGE(TIME:0, 480, 1020) == 0
	PRINTFORMW 영업시간 외 같다
	RETURN -1
ENDIF

;初回イベント
IF FLAG:마지막가구점방문일 == 0
	PRINTL
	PRINTFORMW 가구점、と書かれた看板が目に留まり、軒をくぐってはみたものの…
	PRINTFORMW 「……んだよ、そのツラは。あぁん？」
	PRINTFORMW 「あ？　そうだよ、ここは가구점だよ。なに？　売り物はないのかって？」
	PRINTFORMW 「アホぬかせ。タンスにしろ何にしろ、注文受けてからこさえるもんだ。で、今は暇してるだけの話だよ」
	PRINTFORMW 「……それで？　何が欲しいんだ、アンタ」
	PRINTFORMW 「……あぁ？　実物を見てから決めようと思ってた？」
	PRINTFORMW 「冷やかしなら帰んな……と言いたいとこだが、しゃあねえ。えぇと、前の仕事の帳面はっと――」
	PRINTFORMW 　男が、ずい、と突き出してきた書面には、存外に几帳面そうな字で費用が書かれていたが
	PRINTFORMW 　……うわ、高ッ！？
	PRINTFORMW 「そりゃあ腕によりをかけた逸品揃いなんだ、そんな安請け合いするわけねェだろ」
	PRINTFORMW 　そこを何とかなりませんかねー、と尋ねてみると…
	PRINTFORMW 「金に胡坐を掻いて、一から十までこっちにブン投げるんなら、思いっきり吹っ掛けてやるトコだが」
	PRINTFORMW 「……そうだな。材料になる『製材』を用意してくれりゃ、その分値引いて受けてやってもいいぜ」
	PRINTFORMW 　これこれこれくらい、と、額面を示される……まあ、何とか払えそうな額面に落ち着いていた
	PRINTFORMW 「ふん。まぁ、用意ができたら発注してくんな。最高の逸品を仕上げてやるよ」
	PRINTL
	루프수 = 1
ENDIF

$LOOP_SELECT

IF 루프수
	PRINTL
	IF FLAG:가구작성잔여
		PRINTFORMW 「どいたどいたぁ、俺ぁ忙しいんだ！」
	ELSE
		PRINTFORMW 「なんだ、まだなんか用かよ」
	ENDIF
ELSE
	;平常来店
	PRINTL
	;가구発注中
	IF FLAG:가구작성잔여 >= 14
		PRINTFORMW 「なんだ、アンタか。気が早ェよ、まだ図面引いたばっかだっての」
	ELSEIF FLAG:가구작성잔여 >= 9
		PRINTFORMW 「なんだ、アンタか。注文の가구ならまだまだかかるぜ」
	ELSEIF FLAG:가구작성잔여 >= 5
		PRINTFORMW 「なんだ、アンタか。注文の가구なら進んでるよ、心配すんな」
	ELSEIF FLAG:가구작성잔여 >= 2
		PRINTFORMW 「おう、もうちょいで注文の가구、仕上がるぜ！」
	ELSEIF FLAG:작성중가구 && FLAG:가구작성잔여 == 0
		;完成日
		IF FLAG:작성중가구 == 902
			PRINTFORMW 「よお、よく来たな。　え？　風呂の完成は近くで見届けたし、なんでわざわざ店まで来なきゃなんないのか、だと？」
			PRINTFORMW 「納品書の取り交わしとか色々あんだよ。その辺、きっちりやらないと後腐れが残るのも嫌なんだ」
		ELSE
			PRINTFORMW 「よお。待たせたな。これが、注文の品だ」
		ENDIF
		IF FLAG:작성중가구 == 907
			PRINTFORMW 「いや……まぁ、流石にホウキは専門じゃないからな、ちょっと手間がかかったぜ」
		ELSEIF ITEM:(FLAG:작성중가구) == 2
			PRINTFORMW 「いやぁ……久々に納得の仕事だった。実に、職人冥利に尽きるってもんだ」
		ELSEIF ITEM:(FLAG:작성중가구) == 1
			PRINTFORMW 「ふん、気になったところは手を入れたが……まあ、こんなもんだろ」
			PRINTFORMW 「作る側の満足と、使う側の満足は違う微妙にズレた所にあるが、それを両立するのが職人ってもんだからな」
		ELSE
			PRINTFORMW 「欲しい가구がありゃ、また声をかけろよ。あぁ、今の가구に不満を感じたら、それも教えてくれ」
		ENDIF
		PRINTL
		SETCOLOR 0x00FF00
		PRINTFORM %ITEMNAME:(FLAG:작성중가구)%
		IF ITEM:(FLAG:작성중가구) == 0
			PRINTFORMW を手に入れた！
		ELSE
			PRINTFORMW が改修され、使い心地が良くなった！
		ENDIF
		RESETCOLOR
		PRINTL
		ITEM:(FLAG:작성중가구) ++
		FLAG:작성중가구 = 0
	ELSE
		IF DAY:0 - FLAG:마지막가구점방문일 > 28
			PRINTFORMW 「なんだ、アンタか。いよいよ注文か？」
		ELSEIF FLAG:마지막가구점방문일 == DAY:0
			PRINTFORMW 「なんだよ。アンタに構ってられるほど………ああ、暇だよ、悪ぃか」
		ELSE
			PRINTFORMW 「なんだ、アンタか。가구が欲しいのか？」
		ENDIF
	ENDIF
ENDIF

루프수 ++
FLAG:마지막가구점방문일 = DAY:0

PRINTFORM 　　소지금: %TOSTR(MONEY,"#,###")%
IF FLAG:작성중가구
	PRINTFORML 　　%ITEMNAME:(FLAG:작성중가구)%を作成中　納期：あと{FLAG:가구작성잔여}日
ELSE
	PRINTL
ENDIF
DRAWLINE

CALL ASK_M("가구を作ってほしい", !FLAG:작성중가구, "取引内容を確認する", 1, "店を出る", 2)

SELECTCASE RESULT
	;가구を作ってほしい
	CASE 0
		PRINTFORMW 「アンタが必要そうな가구となると……まぁ、こんなところか？」

		$INPUT_LOOP01
		;リストの呼び出し
		삭제행수 = 0
		FOR LOCAL, 902, 908
			삭제행수 ++
			IF ITEM:LOCAL == 3
				PRINTPLAINFORM 　[{LOCAL}]
				LOCALS = %ITEMNAME:LOCAL%Lv{ITEM:LOCAL}
				PRINTPLAINFORM %LOCALS, 21, LEFT%
			ELSE
				PRINTFORM
				PRINTFORM [{LOCAL}]
				IF ITEM:LOCAL == 0
					;未所持
					PRINTFORM %ITEMNAME:LOCAL, 21, LEFT%
				ELSE
					;所持
					LOCALS = %ITEMNAME:LOCAL%Lv{ITEM:LOCAL}
					PRINTFORM %LOCALS, 21, LEFT%
				ENDIF
			ENDIF
			CALL FURNITURE_EXPLANATION(LOCAL)
		NEXT
		PRINTFORML 　[999] 역시 그만둔다
		삭제행수 ++

		INPUT

		결과 = RESULT
		MaterialQTY = FURNITURE(결과, "軽減上限", ITEM:결과)
		SELECTCASE 결과
			CASE 902 TO 907
				;Lv3の時は選択できない
				IF ITEM:결과 == 3
					CLEARLINE 삭제행수 + 1
					GOTO INPUT_LOOP01
				ENDIF

				;注文가구の請求を作成
				DRAWLINE
				PRINTFORML %ITEMNAME:결과%の\@ ITEM:결과 == 0? 発注# 改修\@
				;所持製材量での値引きの表記
				;Lv1→2の時は半額
				IF ITEM:결과 == 1
					PRINTFORM 　費用　　　　：\\%TOSTR((ITEMPRICE:결과 / 2),"#,###")% →
				ELSE
					PRINTFORM 　費用　　　　：\\%TOSTR(ITEMPRICE:결과,"#,###")%  →
				ENDIF
				DiscountedPrice = %TOSTR(FURNITURE(결과, 0, ITEM:결과),"#,###")%

				IF FURNITURE(결과, 0, ITEM:결과) == FURNITURE(결과, "最低額", ITEM:결과)
					SETCOLOR C_RED
				ELSEIF ITEM:결과 == 1 && FURNITURE(결과, 0, ITEM:결과) < ITEMPRICE:결과 / 2
					SETCOLOR C_YELLOW
				ELSEIF FURNITURE(결과, 0, ITEM:결과) < ITEMPRICE:결과
					SETCOLOR C_YELLOW
				ENDIF
				PRINTFORML \\%DiscountedPrice% \@ FURNITURE(결과, 0, ITEM:결과) == FURNITURE(결과, "最低額", ITEM:결과)? 最安値！# \@
				RESETCOLOR

				;価格軽減に必要な資材量と現在所持量の比較
				PRINTFORM 　価格軽減条件：%WOOD_NAME(FURNITURE(결과, "자재번호"))%の製材PRINTFORM を{MaterialQTY}用意する
				SIF LOG_SALES:FURNITURE(결과, "자재번호") >= MaterialQTY
					SETCOLOR C_RED
				PRINTFORML （{LOG_SALES:FURNITURE(결과, "자재번호")}／{MaterialQTY}）
				RESETCOLOR
			CASE 999
				PRINTL
				GOTO LOOP_SELECT
			CASEELSE
				CLEARLINE 1
				GOTO INPUT_LOOP01
		ENDSELECT
		PRINTL
		IF ITEM:결과 == 0
			CALL ASK_M("この가구を作ってもらう", MONEY >= FURNITURE(결과, 0), "그만둔다", 1)
		ELSE
			CALL ASK_M("この가구を改修してもらう", MONEY >= FURNITURE(결과, 0, ITEM:결과) && ITEM:결과 <= 2, "그만둔다", 1)
		ENDIF

		IF RESULT
			CLEARLINE 9 + 삭제행수
			GOTO INPUT_LOOP01
		ELSE
			;費用の支払い
			PRINTFORMW 　소지금: \\%TOSTR(MONEY,"#,###")% → \\%TOSTR(MONEY - FURNITURE(결과, 0, ITEM:결과),"#,###")%
			PRINTFORM
			CALL PAY_MONEY_YEN(FURNITURE(결과, 0, ITEM:결과))

			;製材等の消費  軽減上限量以上に持っていれば上限量分減らし、上限に達していなければ全部消費
			;現在の製材の量を退避してから計算
			자재보관 = LOG_SALES:FURNITURE(결과, "자재번호")
			LOG_SALES:FURNITURE(결과, "자재번호") = MAX(0, LOG_SALES:FURNITURE(결과, "자재번호") - MaterialQTY)
			;文章部分
			PRINTFORMW 　%WOOD_NAME(FURNITURE(결과, "자재번호"))%の製材：{자재보관} → {LOG_SALES:FURNITURE(결과, "자재번호")}
			PRINTL
			IF ITEM:결과
				PRINTFORMW 「なんだ、前の가구に不満があんのか？　上等じゃねぇか。{FURNITURE(결과, "期間", ITEM:결과)}日後に店に来いよ、ビックリ仰天させてやらぁ」
			ELSE
				PRINTFORMW 「うし、契約成立だ。納期は{FURNITURE(결과, "期間", ITEM:결과)}日後だ。また店に顔出してくんな」
			ENDIF
			FLAG:가구작성잔여 = FURNITURE(결과, "期間", ITEM:결과)
			FLAG:작성중가구 = 결과
		ENDIF
	;取引内容を確認する
	CASE 1
		PRINTFORMW 「あぁ？　もっかい確認したいだとぉ？」
		PRINTFORMW 「こちとら가구점なんだから、가구を作るに決まってんだろ」
		PRINTFORMW 「材料の製材を用意してくれりゃあ、値引きして作ってやるよ」
		PRINTFORMW 「当たり前だが、発注された가구を作るにはそれなりに時間がかかるぞ」
		PRINTFORMW 「ついでに、その期間は注文された가구の作成に専念するから、他の注文は受けないからな」
		PRINTFORMW 「あぁ、もし作った가구に不満があるなら、すぐに言えよ。金は取るが、より上を目指してやんよ」
		PRINTL
	CASE 2
		PRINTFORMW 「さっさと帰んな」
		PRINTL
		TIME += 루프수 * 5
		RETURN -1

ENDSELECT
GOTO LOOP_SELECT

;-------------------------------------------------
;実行可能判定
;-------------------------------------------------
;가구점
@COM_ABLE625
;一括管理
SIF GLOBAL_COMABLE(625)
	RETURN RESULT

IF !AT_HOME(MASTER)
	SIF CFLAG:MASTER:현재위치 != 주택가
		RETURN 0
ELSE
	;長屋前
	SIF CFLAG:MASTER:현재위치 != 214
		RETURN 0
ENDIF
;睡眠中
SIF CFLAG:수면
	RETURN 0
;時間停止中
SIF FLAG:70
	RETURN 0
RETURN 1

;-------------------------------------------------
;注文できる가구リストに付随する説明文
;ARG 가구のアイテム番号
;-------------------------------------------------
@FURNITURE_EXPLANATION(ARG)

IF ITEM:ARG
	PRINTFORM 設置場所：
	SELECTCASE ARG
		CASE 902
			LOCALS = %NAME_FROM_PLACE(NEAR_BATH(1))%
		CASE 904, 907
			LOCALS = %CALLNAME:MASTER%が持ち歩いている
		CASEELSE
			LOCALS = %CALLNAME:MASTER%の私室
	ENDSELECT
	PRINTFORMD %LOCALS, 26, LEFT%
ELSE
	PRINTFORM
ENDIF

SIF ITEM:ARG == 0
	SETCOLOR C_GRAY

SELECTCASE ARG
	CASE 902
		PRINTFORM …設置した場所での入浴の体力・気力回復を増加
	CASE 903
		PRINTFORM …睡眠による体力・気力・精力回復を増加
	CASE 904
		PRINTFORM …料理の出来栄えを良くする
	CASE 905
		PRINTFORM …私室での勉強や調合の効率を上げる
	CASE 906
		PRINTFORM …私室での休憩の体力・気力・精力回復を増加
	CASE 907
		PRINTFORM …掃除の効率を上げる
	CASEELSE
		PRINT
ENDSELECT
IF ITEM:ARG == 3
	SETCOLOR C_ORANGE
	PRINTFORML ▲▲
ELSEIF ITEM:ARG == 2
	SETCOLOR C_YELLOW
	PRINTFORML ▲
ELSE
	PRINTL
ENDIF
RESETCOLOR

;-------------------------------------------------
;가구注文で使用する各種数値をまとめた関数
;現状は902～907
;F_ID …가구のアイテム番号
;ARGS 0= 補正価格(省略時)  1= 資材による軽減上限数
;      2= 가구に対応する자재번호  3= 軽減最低額
;      4= 発注後の作成期間   を返す
;EnhanceTime 0= 初発注(省略時)  1= 強化１  2= 強化２
;        主にITEM:가구番号を入れて利用
;-------------------------------------------------
@FURNITURE(F_ID, ARGS, EnhanceTime = 0)
#FUNCTION
#DIM F_ID
#DIM EnhanceTime
#DIM DiscountRate
#DIM MaxDiscount
#DIM 자재번호
#DIM 軽減上限
#DIM 最低額
#DIM 補正価格
#DIM 期間

;手持ち製材との調整価格を表記
SELECTCASE F_ID
	;ヒノキ風呂
	CASE 902
		자재번호 = 2;ヒノキ
		軽減上限 = 50
		最低額 = 250000
		期間 = 15
	;キリのベッド
	CASE 903
		자재번호 = 8;キリ
		軽減上限 = 30
		最低額 = 200000
		期間 = 10
	;イチョウのまな板
	CASE 904
		자재번호 = 6;イチョウ
		軽減上限 = 10
		最低額 = 50000
		期間 = 5
	;ケヤキの机椅子
	CASE 905
		자재번호 = 7;ケヤキ
		軽減上限 = 30
		最低額 = 200000
		期間 = 10
	;ナラのソファー
	CASE 906
		자재번호 = 9;ナラ
		軽減上限 = 30
		最低額 = 200000
		期間 = 10
	;カスタム竹箒
	CASE 907
		자재번호 = 12;竹
		軽減上限 = 30
		最低額 = 30000
		期間 = 10
ENDSELECT

;Lv1→Lv2時は半分  Lv2→Lv3時は最初と同じで
MaxDiscount = ITEMPRICE:F_ID - 最低額
IF EnhanceTime == 1
	軽減上限 /= 2
	最低額 /= 2
	期間 /= 2
	MaxDiscount /= 2
ENDIF

SIF ARGS == "자재번호"
	RETURNF 자재번호
SIF ARGS == "軽減上限"
	RETURNF 軽減上限
SIF ARGS == "最低額"
	RETURNF 最低額
SIF ARGS == "期間"
	RETURNF 期間

DiscountRate = 100 * MIN(LOG_SALES:자재번호, 軽減上限) / 軽減上限

IF EnhanceTime == 1
	補正価格 = (ITEMPRICE:F_ID / 2) - MaxDiscount * DiscountRate / 100
ELSE
	補正価格 = ITEMPRICE:F_ID - MaxDiscount * DiscountRate / 100
ENDIF
RETURNF 補正価格

path: ERB/커맨드관련/COMF/외출계/COMF625 가구점.ERB


code:
@COM626
#DIM LPCOUNT
#DIM 상품상한
#DIM GIFT_LIST, 11
#DIM GIFT_PRICE, 11
#DIMS GIFT_NAME, 11
#DIM BOUGHT_GIFT
#DIM SCORE
#DIM 생성완료
#DIM 재로드수
#DIM 호감도계수
#DIM 받은횟수
#DIM 건네준횟수

;とりあえず５つ。11個以上のときは#DIMも修整する
상품상한 = 5

;ソース用の設定
IF CFLAG:TARGET:호감도 < 100
	호감도계수 = 0
ELSEIF CFLAG:TARGET:호감도 < 500
	호감도계수 = 1
ELSEIF CFLAG:TARGET:호감도 < 1000
	호감도계수 = 2
ELSEIF CFLAG:TARGET:호감도 < 5000
	호감도계수 = 3
ELSEIF CFLAG:TARGET:호감도 < 10000
	호감도계수 = 4
ELSEIF CFLAG:TARGET:호감도 < 30000
	호감도계수 = 5
ELSE
	호감도계수 = 6
ENDIF

PRINTL
DRAWLINE
;店員に喋らせる場合（例文）
IF 0
	PRINTFORML 「어서오세요」
ELSE
	PRINTFORML %CALLNAME:TARGET%에게 뭔가 사주자……
ENDIF
DRAWLINE

FOR LPCOUNT, 1, 상품상한 + 1
	;品出し,通るたびにラインナップが変わるので注意
	IF !생성완료
		;地域制限を取っ払い全てのパターンを生成する場合
		IF 0
			CALL CREATE_GIFT(99)
		ELSE
			CALL CREATE_GIFT(CFLAG:MASTER:데이트중)
		ENDIF
		GIFT_LIST:LPCOUNT = RESULT:0
		GIFT_PRICE:LPCOUNT = RESULT:1
		GIFT_NAME:LPCOUNT = %RESULTS%
	ENDIF
	PRINTFORML [{LPCOUNT, 2}] - %GIFT_NAME:LPCOUNT, 36, LEFT%\\{GIFT_PRICE:LPCOUNT}
NEXT

DRAWLINE
PRINTFORML 소지금:\\{MONEY}
SIF 재로드수 == 8
	SETCOLOR C_RED
PRINTFORML [99] - 이외의 상품을 본다 (10분)
RESETCOLOR
PRINTFORML [ 0] - 역시 그만둔다
INPUT

;역시 그만둔다,キャンセルだとリロードできてしまうのでSOURCEまで回す
IF RESULT == 0
	TFLAG:194 = -1
	;ウィンドウショッピングとして少量のソースを獲得
	SOURCE:환락 = 500 + 300 * 호감도계수
;商品リロード
ELSEIF RESULT == 99
	;ひやかしはお断り
	IF 재로드수 >= 8
		TFLAG:194 = -2
		재로드수 = 0
		;閉店ガラガラ（他の店も閉まるのは仕様）
		SETBIT TFLAG:일일일회, 10
	ELSE
		생성완료 = 0
		재로드수 ++
		TIME += 10
		RESTART
	ENDIF
;商品選択
ELSEIF INRANGE(RESULT, 1, 상품상한 + 1)
	;お金足りない
	IF MONEY < GIFT_PRICE:RESULT
		PRINTFORMW 돈이 부족하다……
		생성완료 = 1
		CLEARLINE 9 + 상품상한
		RESTART
	;商品購入
	ELSE
		BOUGHT_GIFT = RESULT
		MONEY -= GIFT_PRICE:BOUGHT_GIFT
		;받은횟수をチェック
		IF !CFLAG:TARGET:선물스톡
			받은횟수 = 1
		ELSE
			받은횟수 = GET_GIFTDATA(CFLAG:TARGET:선물스톡, "횟수") + 1
		ENDIF
		;건네준횟수をチェック
		IF !CFLAG:MASTER:건네준선물
			건네준횟수 = 1
		ELSE
			건네준횟수 = CFLAG:MASTER:건네준선물 / 1000
		ENDIF
		;批評する
		CALL REVIEW_GIFT(TARGET, GIFT_LIST:BOUGHT_GIFT)
		SCORE = RESULT
		[IF_DEBUG]
			PRINTFORML ◆디버그용◆
			PRINTFORML 득점:{SCORE}
			PRINTFORML 득점을 조작할 수 있습니다 (1～999)
			INPUT
			SIF INRANGE(RESULT, 1, 999)
				SCORE = RESULT
		[ENDIF]
		;ストックにGIFT_IDを保存
		CFLAG:TARGET:선물스톡 = 받은횟수 * 10000000000000000 + SCORE * 10000000000000 + DAY * 1000000000 + CFLAG:MASTER:데이트중 * 10000000 + GIFT_LIST:BOUGHT_GIFT
		;TARGETはID,MASTERは渡したキャラを保存
		TCVAR:TARGET:오늘의선물 = CFLAG:TARGET:선물스톡
		TCVAR:MASTER:오늘의선물 = TARGET
		;MASTERは건네준횟수と渡したキャラを保存
		CFLAG:MASTER:건네준선물 = 건네준횟수 * 1000 + TARGET
;		PRINTFORMW ID:{CFLAG:TARGET:선물스톡}
		;ソースの계수
		TFLAG:193 = SCORE / 100

		;ソースの獲得,일일일회限定でコストもかかるのでかなり大きめ（これでも時間効率はスキンシップと同等）
		SOURCE:환락 = 800 + 1500 * TFLAG:193 + 800 * 호감도계수
		SOURCE:수동 = 600 + 1500 * TFLAG:193 + 800 * ABL:순종
		SOURCE:정복 = 600 + 800 * TFLAG:193 + 1500 * ABL:새드끼
		;연인補正
		SIF TALENT:TARGET:연인
			TIMES SOURCE:환락, 2.00
		;祭日補正,評価と二重になるけど
		SIF FESTIVAL() != ""
			TIMES SOURCE:환락, 2.00
		;信頼度,大きめ
		TFLAG:98 = (TFLAG:193 + 1) / 2 + 1

		;地の文用の分岐と기분修整
		IF TFLAG:193 >= 7
			SIF TALENT:TARGET:기분 < 0
				CALL KIGEN_CHANGE(TARGET, 100, 1)
			TFLAG:194 = 3
		ELSEIF TFLAG:193 >= 4
			TFLAG:194 = 2
		ELSE
			TFLAG:194 = 1
		ENDIF

		;일일일회
		SETBIT TFLAG:일일일회, 10
	ENDIF
ELSE
	생성완료 = 1
	RESTART
ENDIF

재로드수 = 0
생성완료 = 0
TIME += 30
RETURN 1

;-------------------------------------------------
;実行可能判定
;-------------------------------------------------
;기념품점
@COM_ABLE626
;一括管理
SIF GLOBAL_COMABLE(626)
	RETURN RESULT
;기념품점ではない場所
SIF !GIFT_SHOP(CFLAG:MASTER:현재위치)
	RETURN 0
;데이트중でない
SIF !FLAG:데이트상대
	RETURN 0
;TARGETが데이트상대ではない
SIF TARGET != FLAG:데이트상대
	RETURN 0
;일일일회
SIF GETBIT(TFLAG:일일일회, 10)
	RETURN 0
;수면중
SIF CFLAG:수면
	RETURN 0
;시간정지중
SIF FLAG:70
	RETURN 0
RETURN 1

@GIFT_SHOP(ARG)
#FUNCTION
;기념품점,（今の所は拠点内ではできない）
;境内,社務所を作りたいところ
SIF ARG == GET_MAP_REPLACEMENT(2)
	RETURNF 1
;境内,寺務所を作りたいところ
SIF ARG == GET_MAP_REPLACEMENT(102)
	RETURNF 1
;集会所,土産物店を作りたいところ
SIF ARG == GET_MAP_REPLACEMENT(219)
	RETURNF 1
;湖畔,正門の方がいいか？
SIF ARG == GET_MAP_REPLACEMENT(343)
	RETURNF 1
;待合室,他にいいところが…
SIF ARG == GET_MAP_REPLACEMENT(403)
	RETURNF 1
;香霖堂
SIF ARG == GET_MAP_REPLACEMENT(502)
	RETURNF 1
;中有の道
SIF ARG == GET_MAP_REPLACEMENT(601)
	RETURNF 1
;山道・沢へ続く道,休憩所の方がいいか？
SIF ARG == GET_MAP_REPLACEMENT(702)
	RETURNF 1
;境内,社務所を
SIF ARG == GET_MAP_REPLACEMENT(811)
	RETURNF 1
;大通り
SIF ARG == GET_MAP_REPLACEMENT(931)
	RETURNF 1
;玉兎の街
SIF ARG == GET_MAP_REPLACEMENT(1070)
	RETURNF 1
RETURNF 0

path: ERB/커맨드관련/COMF/외출계/COMF626 기념품점.ERB


code:
;-------------------------------------------------
;홉고블린에의뢰
;외출 커맨드
;-------------------------------------------------
@COM630
PRINTFORML 홉고블린에게 뭔가 부탁하겠습니까?
CALL ASK_M("부탁하지 않는다", 1,"청소 대행",!TFLAG:출장청소,"물주기대행", ITEM:150)
SIF !RESULT
	RETURN -1
IF RESULT == 1
	PRINTFORML 홉고블린에게 %GET_MAPNAME(MAIN_MAP)%의 청소를 부탁하겠습니까?
	CALL PAY_YEN_OR_CM(3000,10)
	SIF !RESULT
		RETURN -1
	PRINTFORML 출장 청소 서비스를 의뢰했습니다.
	TFLAG:출장청소 = 1
ELSEIF RESULT == 2
	PRINTFORML 홉고블린에게 텃밭의 물주기를 부탁하겠습니까? (이번 달까지)
	CALL PAY_YEN_OR_CM(15000,70)
	SIF !RESULT
		RETURN -1
	PRINTFORML 물주기대행을 의뢰했습니다.
	FLAG:물주기대행 = 1
ENDIF
TIME += 5
RETURN 1

@COM_ABLE630
;일괄관리
SIF GLOBAL_COMABLE(630)
	RETURN RESULT
SIF MAIN_MAP == 3
	RETURN 0
SIF CFLAG:MASTER:현재위치 != 350
	RETURN 0
;시간정지중
SIF FLAG:70
	RETURN 0
RETURN 1

@HOBGOBLIN(ARG)
#FUNCTION
IF !IN_HOME(ARG)
	SIF ARG == 객실
		RETURNF 1
ENDIF
RETURNF 0

path: ERB/커맨드관련/COMF/외출계/COMF630 홉고블린에게의뢰_t.ERB


code:
;-------------------------------------------------
;홍마카지노
;외출 커맨드
;-------------------------------------------------
@COM631
TFLAG:카지노입장시카리스마 = MONEY:2
PRINTFORMW %마스터는% 홍마관내의 카지노 시설로 발길을 옮겼다
CALL CASINO_ENTRANCE

SIF NEMUKE() >= 700
	PRINTFORM 졸음이 한계를 맞이하고 있다
PRINTFORMW %마스터는% 카지노를 뒤로 했다
TIME += 10

@COM_ABLE631
;일괄관리
SIF GLOBAL_COMABLE(631)
	RETURN RESULT
SIF !CASINO_SPOT(CFLAG:MASTER:현재위치)
	RETURN 0
SIF TFLAG:망상중
	RETURN 0
;시간정지중
SIF FLAG:70
	RETURN 0
;시간에 여유가 없다
;활동 한계를 700으로서 그 20분전
SIF NEMUKE() >= 700
	RETURN 0
RETURN 1

@CASINO_SPOT(ARG)
#FUNCTION
;広間
SIF ARG == GET_MAP_REPLACEMENT(309)
	RETURNF 1
RETURNF 0

path: ERB/커맨드관련/COMF/외출계/COMF631 홍마카지노_t.ERB


code:
;-------------------------------------------------
;대련
;외출 커맨드
;-------------------------------------------------
@COM632
PRINTFORML 전투의 체험판입니다. 해보겠습니까?
CALL ASK_M("안 한다", 1,"엄청 약한 적", 1,"약한 적", 1)
IF !RESULT
	RETURN -1
ELSE
	CALL BATTLE_SYSTEM(RESULT)
	IF RESULT == -1
		PRINTFORMW 졌다!
	ELSEIF RESULT == 1
		PRINTFORMW 이겼다! 장하다!
	ENDIF
ENDIF

@COM_ABLE632
;일괄관리
SIF GLOBAL_COMABLE(632)
	RETURN RESULT
SIF !TFLAG:100
	RETURN 0
;정지 중에는 불가
SIF FLAG:70 == 1
	RETURN 0
;업무중
SIF CFLAG:행동 == 5
	RETURN 0
SIF !전투훈련가능(CFLAG:MASTER:현재위치)
	RETURN 0
;기력 0
SIF BASE:MASTER:기력 <= 0
	RETURN 0
SIF CFLAG:우후후 == 2
	RETURN 0
RETURN 1

path: ERB/커맨드관련/COMF/외출계/COMF632 대련_t.ERB


code:
@COM650
PRINTL 「git도 사용할 수 없는 사람에게 파는 것은 없어」
PRINTW git는 상식의 범주에 들어가는 것 같다
RETURN -1

path: ERB/커맨드관련/COMF/외출계/COMF650 git_t.ERB


code:
;-------------------------------------------------
;방을방문한다
;외출 커맨드
;-------------------------------------------------
@COM697
#DIM 집주인
#DIM 질투
#DIM 바람기
#DIM 방의더러움
FOR LOCAL, 1, CHARANUM
	IF CFLAG:LOCAL:자택위치 == CFLAG:MASTER:현재위치
		SIF CFLAG:LOCAL:출금
			CONTINUE
		SIF LOCAL == TARGET
			CONTINUE
		SIF CFLAG:LOCAL:약점잡힘
			SETCOLOR C_AQUA
		SIF CFLAG:LOCAL:현재위치 != SUKIMA() || !VISIT(LOCAL) || CFLAG:LOCAL:거점내방 || BASE:LOCAL:체력 < 1000
			SETCOLOR C_GRAY
		PRINTFORML [{LOCAL}] %OMANEKI_PLACE(LOCAL)%
		RESETCOLOR
	ENDIF
NEXT
DRAWLINE
PRINTL [999] 그만둔다
INPUT
SIF RESULT == 999
	RETURN -1
SIF CFLAG:RESULT:자택위치 != CFLAG:MASTER:현재위치
	RETURN -1

집주인 = RESULT

;다른 아이와 데이트중
IF FLAG:데이트상대
	PRINTFORMW %OMANEKI_PLACE(집주인)%에 들어가려 했으나, %CALLNAME:(FLAG:데이트상대)%에게 마음껏 노려보였다
	RETURN -1
ENDIF

;볼 수 있지 않은가 체크
바람기 = 0
질투 = 0
FOR LOCAL, 1, CHARANUM
	IF LOCAL != 집주인 && CFLAG:LOCAL:현재위치 == CFLAG:MASTER:현재위치
		SIF LOCAL == 60
			질투 = 60
		SIF TALENT:LOCAL:연모 && CFLAG:LOCAL:신뢰도 < 1000
			바람기 = LOCAL
	ENDIF
NEXT
;파르시가 출현해 4명이나 살고 있는 지령전은 꽤 난이도가 높다
;연동 캐릭터도 달려 들어 오므로 양다리 하고 있다고(면) 큰 일
IF 질투 || 바람기
	PRINTFORM %OMANEKI_PLACE(집주인)%에 들어가려 했으나,
	IF 질투
		PRINTFORMW 오니같은 형상의 %CALLNAME:질투%에게 방해받았다
	ELSE
		PRINTFORMW 의아한 얼굴을 한 %CALLNAME:바람기%에게 불러세워졌다
	ENDIF
	RETURN -1
ENDIF

IF TCVAR:집주인:댁방문
	PRINTFORML 오늘은 이제 그만 두자
	RETURN -1
ENDIF

SELECTCASE RAND:10
	CASE IS <= 1
		TFLAG:193 = 1
	CASE 2 TO 4
	;매우 바쁜
		TFLAG:193 = 2
	CASE 5,6,7
	;잠시
		TFLAG:193 = 3
	CASE 8,9
	;올려 줄 수 있다
		TFLAG:193 = 4
ENDSELECT
SIF !TALENT:집주인:연모 && !TALENT:집주인:사모 && TFLAG:193 > 2
	TFLAG:193 --

SIF CFLAG:집주인:현재위치 != SUKIMA() || !VISIT(집주인) || CFLAG:집주인:거점내방 || BASE:집주인:체력 < 1000
	TFLAG:193 = 1

PRINTFORMW %마스터는% %조사처리(OMANEKI_PLACE(집주인),"를")% 방문했다
TCVAR:집주인:댁방문 = 1
[IF_DEBUG]
IF FLAG:테스트용딸딸이들킴발생
	CALL ONABARE(집주인)
	GOTO OWARI
ENDIF
[ENDIF]

CALL IRAI_VISIT_HOME(집주인)
SIF RESULT
	GOTO OWARI

IF DAY:2 == 3 && (DAY:3 == 13 || DAY:3 == 15) && TFLAG:193 > 1
	CFLAG:집주인:마지막에만난날 = DAY
	PRINTFORML %조사처리(CALLNAME:집주인,"가")% 나왔다
	PRINTFORMW 「복조리 받아라, 밭에 풍년 받아라!」%마스터는% 풍년을 빌었다
	IF ABL:집주인:교양 >= 5 || 집주인 == 77 || 집주인 == 78
		PRINTFORML %조사처리(CALLNAME:집주인,"는")% 과자를 주었다!
		;현상 실장되고 있는 과자를 랜덤에
		SELECTCASE RAND:5
			CASE 0
				ITEM:120 ++
			CASE 1
				ITEM:212 ++
			CASE 2
				ITEM:213 ++
			CASE 3
				ITEM:216 ++
			CASEELSE
				ITEM:376 ++
		ENDSELECT
	ELSE
		PRINTFORML %조사처리(CALLNAME:집주인,"는")% 의아한 얼굴을 하고 있다…아무래도 「복조리」를 모르는 것 같다
		PRINTFORML %조사처리(CALLNAME:집주인,"는")% 말없이 문을 닫아버렸다
		PRINTFORML %마스터는% 슬퍼졌다
	ENDIF
	TIME += 10
	RETURN 1
ENDIF

IF TFLAG:193 > 1 && CFLAG:집주인:욕구불만도 > 300
	IF RAND:(CFLAG:집주인:욕구불만도 + TFLAG:행운보정 * 50) >= RAND:700
		CALL ONABARE(집주인)
		GOTO OWARI
	ELSEIF ITEM:토부「개운대문」
		PRINTL 이런, 개운대문의 모습이…?
		CALL ASK_YN("꺼낸다","방치한다")
		IF !RESULT
			ITEM:토부「개운대문」 --
			CALL COLORMESSAGE(@"토부「개운대문」이 발동!", C_YELLOW, 2)
			CALL ONABARE(집주인)
			GOTO OWARI
		ENDIF
	ENDIF
ENDIF
SELECTCASE TFLAG:193
	CASE 1
		IF CFLAG:MASTER:현재위치 == CFLAG:집주인:현재위치
			PRINTFORMW 눈 앞에 있는 %조사처리(CALLNAME:집주인,"가")%,  멋대로 들어가지말라고 했다
		ELSE
			PRINTFORMW 대답이 없다… 부재중이거나 자고 있을 것이다
		ENDIF
	CASE 2
		CFLAG:집주인:마지막에만난날 = DAY
		PRINTFORMW %조사처리(CALLNAME:집주인,"가")% 나왔다
		IF !TALENT:집주인:연모 && !TALENT:집주인:사모
			PRINTFORM %조사처리(CALLNAME:집주인,"는")% 귀찮은 듯,
		ELSE
			PRINTFORM %조사처리(CALLNAME:집주인,"는")% 미안한 듯,
		ENDIF
		PRINTFORMW 오늘은 시간이 없으니까 또 다음에라고 말했다
		PRINTFORM 내일 만날 수 있는지 물어보니까,
		IF !RAND:2 && !GETBIT (TFLAG:일일일회, 5)
			PRINTFORMW 내일은 %GET_MAPNAME(MAIN_MAP)%에 갈 생각이라는 것이다
			CFLAG:집주인:다음날내방 = 1
			SETBIT TFLAG:일일일회, 5
		ELSE
			PRINTFORMW 좀 모르겠다, 라는 말을 들었다
		ENDIF
		CALL COM697_1(집주인)
		TIME += 5
	CASE 3
		CFLAG:집주인:마지막에만난날 = DAY
		PRINTFORML %조사처리(CALLNAME:집주인,"가")% 나왔다
		PRINTFORML %조사처리(CALLNAME:집주인,"를")% 보고 싶어져서 왔다고 하자, 조금이라면 라며 들여보내 주었다
		PRINTFORMW …
		PRINTFORMW ……
		PRINTFORMW ………
		PRINTFORMW %조사처리(CALLNAME:집주인,"와")% 잠시 담소했다
		PRINTFORMW 헤어질 때, 내일 만날 수 있는지 물어보았다
		IF !RAND:2 && !GETBIT (TFLAG:일일일회, 5)
			PRINTFORMW 내일은 %GET_MAPNAME(MAIN_MAP)%에 갈 생각이라는 것이다
			CFLAG:집주인:다음날내방 = 1
			SETBIT TFLAG:일일일회, 5
		ELSE
			PRINTFORMW 좀 모르겠다, 라는 말을 들었다
		ENDIF
		CALL COM697_1(집주인)
		TIME += 27
	CASE 4
		CFLAG:집주인:마지막에만난날 = DAY
		PRINTFORMW %조사처리(CALLNAME:집주인,"가")% 나왔다
		PRINTFORMW %조사처리(CALLNAME:집주인,"는")% 기쁜 듯하다…
		CALL ASK_YN("데이트를 신청하다","방문한다")
		IF !RESULT || !TALENT:집주인:연모
			IF !RESULT
				PRINTFORMW %CALLNAME:집주인%에게 데이트를 신청했다
			ELSE
				PRINTFORML 방문할 수 있기를 바랐지만, 거부당하고 말았다
				PRINTFORMW 대신 데이트라면 라고 제안되어지자, %마스터는% 두말없이 고개를 끄덕였다
			ENDIF
			CFLAG:MASTER:동행 = 60
			CFLAG:집주인:동행 = 60
			CFLAG:집주인:현재위치 = CFLAG:MASTER:현재위치
			CFLAG:집주인:데이트중 = CFLAG:MASTER:현재위치 / 100
			CFLAG:집주인:거점내방 = 1
			FLAG:데이트상대 = 집주인
			TFLAG:데이트전호감도 = CFLAG:집주인:2
		ELSE
			CFLAG:집주인:데이트중 = CFLAG:MASTER:현재위치 / 100
			CALL OMANEKI_ENTER(집주인)
		ENDIF
ENDSELECT

$OWARI
TIME += 3
RETURN 1

@COM697_1(ARG)
#DIM 환락UP
#DIM 수동UP
#DIM 정복UP

;회화의 콘파치
TFLAG:98 = 5
;고정으로 획득하는 소스
환락UP = 200
수동UP = 0
정복UP = 0
;ABL:친밀을 본다
IF ABL:ARG:친밀 <= 1
	환락UP += (ABL:ARG:순종 * 40)
ELSEIF ABL:ARG:친밀 <= 3
	환락UP += 200 + (ABL:ARG:순종 * 40)
ELSEIF ABL:ARG:친밀 <= 5
	환락UP += 500 + (ABL:ARG:순종 * 40)
ELSEIF ABL:ARG:친밀 <= 8
	환락UP += 750 + (ABL:ARG:순종 * 60)
ELSEIF ABL:ARG:친밀 <= 10
	환락UP += 1000 + (ABL:ARG:순종 * 60)
ELSE
	환락UP += 1600 + (ABL:ARG:순종 * 30)
ENDIF

;호감 도를 본다
IF CFLAG:ARG:2 <= 1000
	환락UP += CFLAG:ARG:2
ELSEIF CFLAG:2 <= 5000
	환락UP += 500 + (CFLAG:ARG:2 - 500) / 3
ELSE
	환락UP += 2000 + (CFLAG:ARG:2 - 5000) / 5
ENDIF
SIF 환락UP < 0
	환락UP = 0

수동UP = 100 + 100 * ABL:ARG:순종
정복UP = 100 + 100 * ABL:ARG:새드끼

SELECTCASE ABL:MASTER:화술기능
	CASE 0
		TIMES 환락UP , 0.20
		TIMES 정복UP , 0.20
		TIMES 수동UP , 0.20
	CASE 1
		TIMES 환락UP , 0.40
		TIMES 정복UP , 0.40
		TIMES 수동UP , 0.40
	CASE 2
		TIMES 환락UP , 0.70
		TIMES 정복UP , 0.70
		TIMES 수동UP , 0.70
	CASE 3
		TIMES 환락UP , 1.00
		TIMES 정복UP , 1.00
		TIMES 수동UP , 1.00
	CASE 4
		TIMES 환락UP , 1.20
		TIMES 정복UP , 1.20
		TIMES 수동UP , 1.20
	CASEELSE
		TIMES 환락UP , 1.50
		TIMES 정복UP , 1.50
		TIMES 수동UP , 1.50
ENDSELECT

IF TFLAG:193 == 3
	TIMES 환락UP , 3.50
	TIMES 정복UP , 3.50
	TIMES 수동UP , 3.50
	TFLAG:호감도보너스 = 200
ENDIF
SOURCE:ARG:환락 = 환락UP
SOURCE:ARG:정복 = 정복UP
SOURCE:ARG:수동 = 수동UP

CALL FAVOR_CALC(ARG)
CALL TRUST_CALC(ARG)
TFLAG:호감도보너스 = 0

;-------------------------------------------------
;실행 가능 판정
;-------------------------------------------------
;방을방문한다
@COM_ABLE697
;일괄관리
SIF GLOBAL_COMABLE(697)
	RETURN RESULT
SIF AT_HOME(MASTER)
	RETURN 0
;시간정지중
SIF FLAG:70
	RETURN 0
LOCAL:1 = 0
FOR LOCAL, 1, CHARANUM
	SIF CFLAG:LOCAL:출금
		CONTINUE
	SIF CFLAG:LOCAL:자택위치 == CFLAG:MASTER:현재위치 && !CFLAG:LOCAL:신사거주 && LOCAL != TARGET; && (TALENT:LOCAL:사모 || TALENT:LOCAL:연모)
		LOCAL:1 ++
NEXT

SIF !LOCAL:1
	RETURN 0
RETURN 1

path: ERB/커맨드관련/COMF/외출계/COMF697 방을방문한다_t.ERB


code:
;-------------------------------------------------
;방에들어간다
;외출 커맨드
;-------------------------------------------------
@COM698
#DIM 방의더러움
#DIM 바람기
#DIM 질투

;다른 아이와 데이트중
IF FLAG:데이트상대 != TARGET && FLAG:데이트상대 != 0
	PRINTFORM %OMANEKI_PLACE()%에 들어가려 했으나, %CALLNAME:(FLAG:데이트상대)%에게 마음껏 노려보였다
	RETURN -1
ENDIF

;볼 수 있지 않은가 체크
바람기 = 0
질투 = 0
FOR LOCAL, 1, CHARANUM
	IF LOCAL != TARGET && CFLAG:LOCAL:현재위치 == CFLAG:MASTER:현재위치
		SIF LOCAL == 60
			질투 = 60
		SIF TALENT:LOCAL:연모 && CFLAG:LOCAL:신뢰도 < 1000
			바람기 = LOCAL
	ENDIF
NEXT
;파르시가 출현해 4명이나 살고 있는 지령전은 꽤 난이도가 높다
;연동 캐릭터도 달려 들어 오므로 양다리 하고 있다고(면) 큰 일
IF 질투 || 바람기
	PRINTFORM %OMANEKI_PLACE(집주인)%에 들어가려 했으나,
	IF 질투
		PRINTFORMW 오니같은 형상의 %CALLNAME:질투%에게 방해받았다
	ELSE
		PRINTFORMW 의아한 얼굴을 한 %CALLNAME:바람기%에게 불러세워졌다
	ENDIF
	RETURN -1
ENDIF

CALL OMANEKI_ENTER(TARGET)

RETURN 1

;-------------------------------------------------
;실행 가능 판정
;-------------------------------------------------
;방에들어간다
@COM_ABLE698

;일괄관리
SIF GLOBAL_COMABLE(698)
	RETURN RESULT
;업무중
SIF CFLAG:행동 == 5
	RETURN 0
;자택에 없다
SIF CFLAG:MASTER:현재위치 != CFLAG:자택위치
	RETURN 0
;우선 연모만 조건
SIF !TALENT:연모
	RETURN 0
;외출하고 있지 않다
SIF !CHK_DATENOW(CFLAG:MASTER:데이트중)
	RETURN 0
;다른 아이와 데이트중, 재미있기 때문에 커맨드는 실행할 수 있도록(듯이)
;SIF FLAG:데이트상대 != TARGET && FLAG:데이트상대 != 0
;	RETURN 0
;화내고 있으면(자) 안 돼(버그 회피 용무라도 있다)
SIF BASE:분노 > 50
	RETURN 0
;수면중
SIF CFLAG:수면
	RETURN 0
;시간정지중
SIF FLAG:70
	RETURN 0
RETURN 1

;-------------------------------------------------
;외박 처리
;-------------------------------------------------
@OMANEKI_ENTER(ARG)
#DIM 방의더러움

;초대 플래그에 ARG를 대입
CFLAG:ARG:초대 = ARG
CFLAG:MASTER:초대 = ARG

;초기위치를 OMANEKIBEYA()=98로 한다
CFLAG:ARG:초기위치 = OMANEKIBEYA()
;이동시킨다
CFLAG:ARG:현재위치 = OMANEKIBEYA()
CFLAG:MASTER:현재위치 = OMANEKIBEYA()

;방이 더러워지는 상태
;한 번 들어가 있으면(자) 참조한다
IF TCVAR:ARG:초대방의더러움 > 0
	YOGORE:(OMANEKIBEYA()) = TCVAR:ARG:초대방의더러움
ELSE
	;최초로 들어갈 때는 악취내성 의존(민감 이하=최고, 보통=중, 둔감=저, 무시=쓰레기산)
	방의더러움 = MAX(1999 + (TALENT:ARG:61 * 2000), 0)
	YOGORE:(OMANEKIBEYA()) = 방의더러움
	TCVAR:ARG:초대방의더러움 = 방의더러움
ENDIF

path: ERB/커맨드관련/COMF/외출계/COMF698 방에들어간다_t.ERB


code:
;-------------------------------------------------
;밖으로나온다
;외출 커맨드
;-------------------------------------------------
@COM699

;PRINTFORMW %STR:(((CFLAG:TARGET:자택위치) / 10) + 6000)%에 나온다

;자택위치의 3자리수째를 참조
CFLAG:MASTER:데이트중 = (CFLAG:(CFLAG:MASTER:초대):자택위치) / 100
;자택위치에 이동
CFLAG:MASTER:현재위치 = CFLAG:(CFLAG:MASTER:초대):자택위치
IF !CFLAG:TARGET:수면
	CFLAG:TARGET:현재위치 = CFLAG:(CFLAG:TARGET:초대):자택위치
ELSE
	CFLAG:TARGET:현재위치 = SUKIMA()
ENDIF
;만남 플래그를 자택위치로 설정(연동하고 있다고(면) 다른 캐릭터가 달려 들어 오는 것은 사양)
CFLAG:(CFLAG:MASTER:초대):만난위치 = CFLAG:(CFLAG:MASTER:초대):자택위치
;방의 더러움을 보존해 둔다
IF YOGORE:(OMANEKIBEYA()) > 0
	TCVAR:(CFLAG:MASTER:초대):초대방의더러움 = YOGORE:(OMANEKIBEYA())
ELSE
	;0 이하로 했을 경우는 식별하기 위해서 1남긴다
	TCVAR:(CFLAG:MASTER:초대):초대방의더러움 = 1
ENDIF
;초대 플래그에 0을 대입
CFLAG:TARGET:초대 = 0
CFLAG:MASTER:초대 = 0

IF CFLAG:TARGET:신사거주
	CFLAG:TARGET:초기위치 = CFLAG:TARGET:신사거주
ELSE
	CFLAG:TARGET:초기위치 = SUKIMA()
ENDIF
;TIME += 3
RETURN 1

;-------------------------------------------------
;실행 가능 판정
;-------------------------------------------------
;밖으로나온다
@COM_ABLE699
;일괄관리
SIF GLOBAL_COMABLE(699)
	RETURN RESULT
;기력 0
SIF BASE:MASTER:기력 <= 0
	RETURN 0
;막차…없어져 버렸다……
SIF NEMUKE() >= 720
	RETURN 0
;초대방전용
SIF CFLAG:MASTER:현재위치 != OMANEKIBEYA()
	RETURN 0
;동침중은 안 되요
SIF CFLAG:MASTER:동침중
	RETURN 0
;장난중도 안 돼
SIF CFLAG:MASTER:장난
	RETURN 0
;우후후 안도 안 돼
SIF CFLAG:MASTER:우후후
	RETURN 0
;시간정지중
SIF FLAG:70
	RETURN 0
RETURN 1

path: ERB/커맨드관련/COMF/외출계/COMF699 밖으로나온다_t.ERB


code:
;釣った魚サイズの記録変数
#DIM SAVEDATA FISHSIZE_RECORD, 60

;釣った魚の数の記録変数
#DIM SAVEDATA FISH_EXP, 60

;材木商に売った材木数の記録変数改め、各製材の所有数を記録する変数
#DIM SAVEDATA LOG_SALES, 50

#DIM WoodValue_Log, 50
#DIM WoodValue_Plank, 50
#DIM WoodEffort, 50
#DIMS WoodName, 50


code:
@ONABARE(ARG)
#DIM 이루어무너뜨려필요치
#DIM 부디그대로호감도상승치
#DIM TARGET보존

CFLAG:ARG:마지막에만난날 = DAY
;EVENT 말을 직접 부르므로 TARGET를 포착
TARGET보존 = TARGET
TARGET = ARG

IF !TALENT:ARG:처녀 && ABL:ARG:A감각 >= ABL:ARG:V감각
	LOCAL:2 = 1
ELSE
	LOCAL:2 = 0
ENDIF

PRINTFORMW \@CFLAG:MASTER:311 == CFLAG:ARG:311? 방 밖에서 모습을 엿봤다# 열쇠가 열려 있어서 마음대로 들어갔다\@
PRINTFORMW 안쪽에서 흐려서 분명하지 않은 목소리가 들린다……

;말체크
TRYCCALLFORM M_KOJO_EVENT_K{ARG}_ONABARE_1(LOCAL:2)

CATCH
	PRINTFORMW 눈치채지 못하도록 방안을 들여다보니, %조사처리(CALLNAME:ARG,"가")% 자위에 빠져 있었다
	PRINTFORM 옷이 벌어지고 아랫배를 드러낸 %조사처리(CALLNAME:ARG,"는")%,
	IF TALENT:ARG:연모
		PRINTFORM %CALLNAME:MASTER%의 이름을 부르
	;심약계, 진면목계
	ELSEIF CFLAG:ARG:성격경향 == 1 || CFLAG:ARG:성격경향 == 4
		PRINTFORM 목소리를 눌러 참으
	;강경세계
	ELSEIF CFLAG:ARG:성격경향 == 2 || CFLAG:ARG:성격경향 == 3 || CFLAG:ARG:성격경향 == 5
		PRINTFORM 달콤한 소리를 지르
	ENDIF
	PRINTFORMW 며, 천천히 손가락을 움직이고 있다

	;큰 가슴
	IF TALENT:ARG:가슴사이즈 > 0
		PRINTFORM 출렁하고 흔들리는 풍만한 유방
	;있고
	ELSEIF TALENT:ARG:가슴사이즈 < 0
		PRINTFORM 부들부들 떨리는 얇은 가슴팍
	;보통
	ELSE
		PRINTFORM 파르르 흔들리는 젖가슴
	ENDIF
	PRINTFORMW 을 어루만지며, 다리를 벌려 비소 쪽으로 손가락을 가져가는 %CALLNAME:ARG%
	PRINTFORMW 그대로, 스윽하며 \@ !LOCAL:2 ? 비소 # 아누스 \@에 가운데 손가락을 찔러넣는 모습에 군침을 삼킨다
	PRINTFORMW 동시에, 쾅하고, 문에 이마가 부딪쳐 버렸다
	PRINTFORML 내방객을 알아챈 %조사처리(CALLNAME:ARG,"는")% 경직되어, 몸을 숨기는 것 조차 잊어버리고 %CALLNAME:MASTER% 쪽을 바라보고 있다……
ENDCATCH

CALL KOJO_MESSAGE_SEND("EVENT", 26, ARG, 0,0)
PRINTFORMW
$LOOP

PRINTFORML 어떻게 하지?
CALL ASK_M("실례했습니다", 1,"봐줄테니까 부디 그대로", 1,"혼자보다는 둘이서 하는 편이 기분이 좋다", BASE:MASTER:정력 > MAXBASE:MASTER:정력 / 5)
IF RESULT == 0
	IF CFLAG:MASTER:311 == CFLAG:ARG:311
		PRINTFORML %마스터는% 허둥지둥 잠자리에 들었다
	ELSE
		PRINTFORML %마스터는% 눈앞의 광경을 망막에 새긴 다음, 동요하는 %조사처리(CALLNAME:ARG,"는")% 정신을 차리기 전에 한마디 사과하고 가버렸다
	ENDIF
	SIF !CFLAG:ARG:약점잡힘
		PRINTFORML %CALLNAME:ARG%의【약점】을 잡았다
	CALL KOJO_MESSAGE_SEND("EVENT", 26, ARG, 1,0)
	CFLAG:ARG:딸딸이들킴 = 1
	CFLAG:ARG:약점잡힘 = 1
	TIME += 5
ELSEIF RESULT == 1
	이루어무너뜨려필요치 = 껴안고필요치(ARG) - 50 + TALENT:ARG:수치심 * 50 - ABL:ARG:노출벽 * 30 - CFLAG:ARG:욕구불만도 / 20
	SIF  TALENT:ARG:자존심 > 0
		이루어무너뜨려필요치 += 100
	SIF CFLAG:ARG:기정사실 & 합의_우후후
		이루어무너뜨려필요치 -= 50
	SIF TALENT:ARG:담력 < 0
		이루어무너뜨려필요치 -= 200
	CALL SHOW_LOVEPOWER(CFLAG:ARG:합의판정, 50)
	CALL SHOW_REASONWALL(이루어무너뜨려필요치 , 50)
	;실패
	IF CFLAG:ARG:합의판정 <= 이루어무너뜨려필요치
		GOTO GOOUT
	;성공
	ELSE
		CALL KOJO_MESSAGE_SEND("EVENT", 26, ARG, 2,0)
		;말체크
		TRYCCALLFORM M_KOJO_EVENT_K{NO:ARG}_ONABARE_2(LOCAL:2)
		CATCH
			PRINTFORM 잠시
			IF TALENT:ARG:연모
				PRINTFORM 기쁨과 당황스러움을 나타내
			;심약계, 진면목계
			ELSEIF CFLAG:ARG:성격경향 == 1 || CFLAG:ARG:성격경향 == 4
				PRINTFORM 머뭇머뭇 쑥스러운 듯하
			;강경세계, 솔직함계, 그 외
			ELSEIF CFLAG:ARG:성격경향 == 2 || CFLAG:ARG:성격경향 == 3 || CFLAG:ARG:성격경향 == 5
				PRINTFORM 어이없다는 듯 난처한 듯한 표정을 짓
			ENDIF
			PRINTFORML 고 있던 %조사처리(CALLNAME:ARG,"는")%, 작게 한숨을 쉬었다
			IF TALENT:ARG:연인
				PRINTFORML 양어께의 피부에 옷을 벗어던지고, %마스터를% 정면으로 마주하며 크게 다리를 벌렸다
			;심약계, 진면목계
			ELSEIF CFLAG:ARG:성격경향 == 1 || CFLAG:ARG:성격경향 == 4
				PRINTFORML 부끄러운 듯이 옷만 조금 풀어헤쳤을 뿐, 다리는 여전히 닫혀 있다
			;강경세계, 솔직함계, 그 외
			ELSEIF CFLAG:ARG:성격경향 == 2 || CFLAG:ARG:성격경향 == 3 || CFLAG:ARG:성격경향 == 5
				PRINTFORML 옷을 벗으면서, 그래도 중요한 부분은 손으로 가리고 있다
			ENDIF
			PRINTFORML 그런, %CALLNAME:ARG%에게, 계속을 재촉하는 %CALLNAME:MASTER%
			PRINTFORMW 작게 한숨을 내쉰 %조사처리(CALLNAME:ARG,"가")% 천천히, 손을 움직이기 시작했다

			PRINTFORML 찌꺽찌꺽하며 소리를 내며 \@ !LOCAL:2 ? 질구 # 항문 \@에 파묻히는 가운데 손가락
			;큰 가슴
			IF TALENT:가슴사이즈 > 0
				PRINTFORML 폭신부들 손가락이 파묻혀 모양을 바꾸는 젖가슴
			;있고
			ELSEIF TALENT:가슴사이즈 < 0
				PRINTFORML 빙글빙글 손가락 끝에 문지러 세워지는 젖꼭지
			;보통
			ELSE
				PRINTFORML 폭신폭신 손으로 흔들리고 있는 젖가슴
			ENDIF
			PRINTFORMW 그 선정적인 광경을 더욱 음란하게 하는 달콤한 목소리가, 울리기 시작했다
			IF TALENT:ARG:연인
				PRINTFORML 넉살좋게, %CALLNAME:MASTER%의 이름을 부르면서, 격렬하게 손가락을 움직이기 시작했다
				PRINTFORML 끈적 끈적하게, 위와 아래 입에서 침을 흘리고, 황홀한 표정을 지어오는 %CALLNAME:ARG%
				PRINTFORMW 꿀꺽하고 침을 삼키는 %CALLNAME:MASTER%
				PRINTFORML 서서히 가슴을 주무르는 손과, \@ !LOCAL:2 ? 질구 # 항문 \@에 물려진 손가락의 움직임이 빨라져 간다
				PRINTFORMW 그리고, %CALLNAME:MASTER%의 이름을 부르면서, 온몸을 떨며 절정했다
			;심약계, 진면목계
			ELSEIF CFLAG:ARG:성격경향 == 1 || CFLAG:ARG:성격경향 == 4
				PRINTFORML 부끄러운 듯, 가슴과 \@ !LOCAL:2 ? 비소 # 아누스 \@를 만지작거리는 %CALLNAME:ARG%
				PRINTFORML 하지만, 희미하게 혈색이 좋아지는 피부에, 눈에 보여져 날카로워지는 젖꼭지에, 열락이 엿보인다
				PRINTFORMW 점점 목소리가 커지고, 조금씩 다리가 벌어진다
				PRINTFORML %CALLNAME:MASTER%의 시선을 눈치채지 못한 듯, 조심스럽지만 감출 수 없을 정도로 다리가 벌어졌다
				PRINTFORML 주르륵, 애액이 늘어지고, 동시에, 움찔하고 허벅지가 경련한다
				PRINTFORMW 너무나 갑작스럽게 애액을 뿜으면서, %조사처리(CALLNAME:ARG,"는")% 끝에 다다랐다
			;강경세계, 솔직함계, 그 외
			ELSEIF CFLAG:ARG:성격경향 == 2 || CFLAG:ARG:성격경향 == 3 || CFLAG:ARG:성격경향 == 5
				PRINTFORML 손을 댈까하고 접근하는 %CALLNAME:MASTER%에게, 보여주기만 한다는 다짐을 해 두는 %CALLNAME:ARG%
				PRINTFORML 가슴을 만지작거리며, 찌꺽 찌꺽 음란한 소리를 내며, \@ !LOCAL:2 ? 비소 # 아누스 \@를 만지작거리고 있다
				PRINTFORMW 이쪽의, 불만스러운 모습을 눈치챘는지, 한숨을 내쉰 %조사처리(CALLNAME:ARG,"가")% 크게 다리를 벌렸다
				PRINTFORML 줄줄 흐르는 애액과, 팽팽하게 날카로워진 클리토리스는 선정적이고
				PRINTFORML 이쪽의 시선을 느끼고 있는지, %CALLNAME:ARG%의 손가락 움직임이 빨라져 간다
				PRINTFORMW 갑자기, \@ !LOCAL:2 ? 질 속 # 항문 속 \@ 깊이 손가락을 찌르고, %조사처리(CALLNAME:ARG,"가")% 달했다
			ENDIF
		ENDCATCH
	ENDIF
	PRINTFORML 실룩실룩, 팔딱팔딱, 허벅지를 경련시켜, 온몸을 진동시키는 %CALLNAME:ARG%
	IF CFLAG:MASTER:311 == CFLAG:ARG:311
		PRINTFORML %마스터는% 탈진한 그녀에게 이불을 덮고 잠자리에 들었다
	ELSE
		PRINTFORML 축 늘어진 그 모습에, 과연 손대기 꺼려진 %마스터는% 그대로 돌아가기로 했다
		IF !CFLAG:ARG:노팬티 && EQUIP:ARG:하반신속옷2 && CFLAG:MASTER:데이트중 != MAIN_MAP
			PRINTFORML 모처럼의 기회이니, 선물을 받아 가시겠습니까?
			CALL ASK_YN
			IF !RESULT
				PRINTFORML 돌아갈 때, %마스터는% %CALLNAME:ARG%의 %조사처리(PANTSNAME(EQUIP:ARG:하반신속옷2, ARG),"를")% 주워 품에 넣었다
				CALL PANTS_GET(ARG)
			ENDIF
		ENDIF
	ENDIF
	CFLAG:ARG:딸딸이들킴 = 3
	부디그대로호감도상승치 = ABL:ARG:친밀 * CFLAG:ARG:욕구불만도 / 250 * MIN(ABL:ARG:노출벽, 10)
	CALL CHANGE_CFLAG(2, ARG, 부디그대로호감도상승치)
	CALL CHANGE_CFLAG(4, ARG, 부디그대로호감도상승치 * -1 / 20)
	CALL JUEL_UP_SPECIAL(ARG,"치정","자위중독", 30,1)
	CALL JUEL_UP_SPECIAL(ARG,"욕정","자위중독", 30,1)
	CALL ADD_EXP_LESNIAN(CFLAG:ARG:욕구불만도 / 100, ARG, "PRINT")
	EXP:ARG:노출쾌락경험 += CFLAG:ARG:욕구불만도 / 200 * ABL:ARG:노출벽
	EXP:ARG:자위경험 +=  CFLAG:ARG:욕구불만도 / 200 * ABL:ARG:노출벽
	TIME += 27
ELSEIF RESULT == 2
	이루어무너뜨려필요치 = 껴안고필요치(ARG) - 150 + TALENT:ARG:수치심 * 30 - CFLAG:ARG:욕구불만도 / 20
	SIF  TALENT:ARG:자존심 > 0
		이루어무너뜨려필요치 += 50
	SIF CFLAG:ARG:기정사실 & 합의_우후후
		이루어무너뜨려필요치 -= 50
	SIF TALENT:ARG:담력 < 0
		이루어무너뜨려필요치 -= 200
	CALL SHOW_LOVEPOWER(CFLAG:ARG:합의판정, 50)
	CALL SHOW_REASONWALL(이루어무너뜨려필요치 , 50)
	$ GOOUT
	IF CFLAG:ARG:합의판정 <= 이루어무너뜨려필요치
		PRINTFORML %조사처리(CALLNAME:ARG,"는")% 알 수 없는 것을 마구 소리쳐 %마스터를% 내쫓았다
		CALL KOJO_MESSAGE_SEND("EVENT", 26, ARG, 3,0)
		SIF !CFLAG:ARG:약점잡힘
			PRINTFORML %CALLNAME:ARG%의【약점】을 잡았다
		CALL GET_ANGRY(ARG, 1)
		CFLAG:ARG:딸딸이들킴 = 1
		CFLAG:ARG:약점잡힘 = 1
		TIME += 5
	ELSE
		CALL NASIKUZUSI(ARG, LOCAL:2)
		CFLAG:ARG:대주고말았다 = 1
	ENDIF
ENDIF
TARGET = TARGET보존

@NASIKUZUSI(ARG, ARG:1)
#DIM 전희1
;1=C 2=B 3=M
#DIM 전희2
#DIM 실전은어느쪽인가
#DIM 주입량
#DIM 기본주입량
#DIM 호감도상승량
#DIM 신뢰저하량
#DIM 회수
#DIM 경험증가수, 58
;0이라면 V, 1이라면 A
VARSET 경험증가수
CALL KOJO_MESSAGE_SEND("EVENT", 26, ARG, 4,0)
회수 = MAX(MIN(ABL:ARG:욕망 / 2, 5) + CFLAG:ARG:욕구불만도 / 200,1)
주입량 = 0
IF TALENT:ARG:키스미경험
	전희1 = 3
	경험증가수:4 += 3 + ABL:ARG:M감각 * 2
	경험증가수:25 = 경험증가수:4
	경험증가수:27 = 경험증가수:4
	EXP:MASTER:M경험 += 경험증가수:4
	EXP:MASTER:키스경험 += 경험증가수:4
	EXP:MASTER:펠라티오경험 += 경험증가수:4
ELSEIF !RAND:3
	전희1 = 3
	경험증가수:4 += 3 + ABL:ARG:M감각 * 2
	경험증가수:27 = 경험증가수:4
ELSEIF ABL:ARG:C감각 >= ABL:ARG:B감각
	전희1 = 1
	경험증가수:0 += RAND:(회수 * 2)
	경험증가수:5 += 경험증가수:0 / 2
	경험증가수:10 += 경험증가수:0 / 2
ELSE
	전희1 = 2
	경험증가수:3 += RAND:(회수 * 2)
	경험증가수:8 += 경험증가수:3 / 2
	경험증가수:10 += 경험증가수:0 / 2
ENDIF
전희2 = RAND:3
실전은어느쪽인가 = ARG:1

SIF 실전은어느쪽인가 == 1
	전희2 += 10
IF 실전은어느쪽인가
	기본주입량 = (ABL:ARG:애널 + 1) * 20 + BASE:MASTER:정력 /25
	경험증가수:21 += 회수
	경험증가수:17 += 회수
	경험증가수:10 += 회수
	경험증가수:7 += 2 * 회수
	경험증가수:2 += 회수 * 2
	EXP:MASTER:삽입경험 += 회수
	MAXBASE:MASTER:정력 += 회수*3;;정력증가
	EXP:MASTER:사정경험 += 회수
	EXP:MASTER:절정경험 += 회수

ELSE
	기본주입량 = (ABL:ARG:질 + 1) * 20 + BASE:MASTER:정력 / 25
	SIF TALENT:ARG:처녀 == 1
		기본주입량 += 100
	경험증가수:20 += 회수
	경험증가수:16 += 회수
	경험증가수:10 += 2 * 회수
	경험증가수:6 += 2 * 회수
	경험증가수:1 += 2 * 회수
	MAXBASE:MASTER:정력 += 회수*3;;정력증가
	EXP:MASTER:삽입경험 += 회수
	EXP:MASTER:사정경험 += 회수
	EXP:MASTER:절정경험 += 회수
ENDIF

FOR LOCAL, 1, 회수
	주입량 += 기본주입량 + RAND:(기본주입량)
NEXT

주입량 = MIN(주입량, BASE:MASTER:정력)

IF !실전은어느쪽인가
	CALL ADD_SAMEN_V(ARG, 주입량)
ELSE
	CALL ADD_SAMEN_A(ARG, 주입량)
ENDIF

;말체크
TRYCCALLFORM M_KOJO_EVENT_K{NO:ARG}_ONABARE_3(전희1, 실전은어느쪽인가, 회수, 주입량)
CATCH
	;이하 Default 지문
	PRINTFORML %마스터는% 동요하는 %조사처리(CALLNAME:ARG,"를")% 뒷전으로 뒷문을 닫아 거침없이 들어가 앉아,
	PRINTFORMW %조사처리(CALLNAME:ARG,"를")% 껴안고 무슨 생각을 하며 하고 있었는지 물었다
	PRINTFORM 그리고 그녀가 원했던 대로,

	SELECTCASE 전희1
		CASE 1
			PRINTDATA
				DATAFORM 클리토리스를 손끝으로 집으며
				DATAFORM 클리토리스를 손가락 끝으로 튕기며
				DATAFORM 음핵을 손가락끝으로 만지작 거리며
				DATAFORM 음핵을 핥으며
			ENDDATA
		CASE 2
			PRINTDATA
				DATAFORM 유두를 손가락으로 집으며
				DATAFORM 가슴을 애무하며
				DATAFORM 유방을 손으로 문질러 비비며
				DATAFORM 유두를 혀로 굴리며
				DATAFORM 유두를 살짝 깨물며
			ENDDATA
		CASE 3
			PRINTDATA
				DATAFORM 머리카락을 쓰다듬고 나서
				DATAFORM 살며시
				DATAFORM 정열적으로
			ENDDATA
	ENDSELECT
	IF 전희1 == 3
		IF TALENT:ARG:키스미경험
			PRINTFORM 첫 키스를 빼앗으며
		ELSE
			PRINTFORM 입을 맞추며
		ENDIF
	ENDIF
	PRINT ,

	SELECTCASE 전희2
		CASE 0
			PRINTFORM 음순 주위를 어루만지며
		CASE 1
			PRINTFORM 질구에 얕게 손가락을 삽입하여
		CASE 2
			PRINTFORM 질구를 혀로 쑤시며
		CASE 10
			PRINTFORM 엉덩이를 차분히 주물러 만지며
		CASE 11
			PRINTFORM 항문 주위를 어루만지며
		CASE 12
			PRINTFORM 항문에 얕게 손가락을 삽입하여
	ENDSELECT
	PRINT ,

	PRINTDATA
		DATAFORM 실컷 애태워 주었다
		DATAFORM 철저히 가게 해 주었다
		DATAFORM 추잡한 말로 괴롭혔다
	ENDDATA
	PRINTW
	PRINTFORM 그리고 %TEXTR("네발로 엎드리게 해서 등뒤에서 짐승처럼 덮고/정면으로 껴안고/굴곡위 자세를 잡게 하고/정면으로 껴안고")%
	PRINTDATA
		DATAFORM 달콤한 말을 속삭이며
		DATAFORM 상냥하게
		DATAFORM 난폭하게
		DATAFORM 끈적끈적
	ENDDATA

	IF TALENT:ARG:처녀 && !실전은어느쪽인가
		PRINTFORM 순결을 빼앗아,
	ELSE
		PRINTFORM {회수}번 %TEXTR("범해/성교해/꿰뚫어")%,
	ENDIF
	PRINTFORMW \@실전은어느쪽인가 ? 애널# 질\@에 {주입량}의 정액을 쏟아 부었다……

	;이상 Default 지문
ENDCATCH

IF 전희1 == 3
	SIF TALENT:MASTER:키스미경험
		CALL SET_FIRSTKISS(MASTER, ARG, 2)
	SIF TALENT:ARG:키스미경험
		CALL SET_FIRSTKISS(ARG, MASTER, 2, "흘러가는 대로")
ENDIF
IF TALENT:ARG:처녀 == -1 && !실전은어느쪽인가
	SETCOLOR 247,171,166
	PRINTFORMW 【무자각비처녀상실】
	TALENT:ARG:처녀 = 0
ENDIF
IF TALENT:ARG:처녀 == 1 && !실전은어느쪽인가
	SETCOLOR 247,171,166
	PRINTFORMW 【처녀상실】
	CALL SET_HISTORY_LOST_V(ARG, MASTER, "페니스", "자기 방", "흘러가는 대로")
ENDIF
IF !CFLAG:ARG:애널처녀상실기념일 && 실전은어느쪽인가
	SETCOLOR 247,171,166
	PRINTFORMW 【애널 처녀상실】
	CALL SET_HISTORY_LOST_A(ARG, MASTER, "페니스", "자기 방", "흘러가는 대로")
ENDIF
SETCOLOR C_YELLOW
FOR LOCAL, 0,58
	IF 경험증가수:LOCAL
		PRINTFORML %EXPNAME:LOCAL% ＋{경험증가수:LOCAL}
		EXP:ARG:LOCAL += 경험증가수:LOCAL
	ENDIF
NEXT
RESETCOLOR
CALL JUEL_UP_SPECIAL(ARG,"욕정","욕망", 20,1)
SELECTCASE 전희1
;C
	CASE 1
		CALL JUEL_UP_SPECIAL(ARG,"쾌C","C감각", 10,1)
;B
	CASE 2
		CALL JUEL_UP_SPECIAL(ARG,"쾌B","B감각", 10,1)
;M
	CASE 3
		CALL JUEL_UP_SPECIAL(ARG,"쾌M","M감각", 20,1)
ENDSELECT
IF 실전은어느쪽인가
	CALL JUEL_UP_SPECIAL(ARG,"쾌A","A감각", 15,1)
ELSEIF TCVAR:ARG:파과
	CALL JUEL_UP_SPECIAL(ARG,"쾌V","V감각", 100,1)
ELSE
	CALL JUEL_UP_SPECIAL(ARG,"쾌V","V감각", 15,1)
ENDIF
CALL ADD_EXP_LESNIAN(회수 * 5, ARG, "PRINT")
PRINTFORMW …
PRINTFORMW ……
PRINTFORMW ………

CALL KOJO_MESSAGE_SEND("EVENT", 26, ARG, 5,0)
TIME += 회수 * 20
EXP:MASTER:청소경험 ++
DOWNBASE:MASTER:체력 = 회수 * 100
DOWNBASE:MASTER:기력 = 회수 * 150
호감도상승량 = 회수 * ABL:ARG:친밀 * 5
신뢰저하량 = 호감도상승량 / 5
SIF TALENT:ARG:사모
	신뢰저하량 /= 2
SIF TALENT:ARG:섹프
	신뢰저하량 /= 2
SIF TALENT:ARG:연모
	신뢰저하량 /= 4

CALL CHANGE_CFLAG(2, ARG, 호감도상승량)
CALL CHANGE_CFLAG(4, ARG, 신뢰저하량 * -1)
CFLAG:ARG:욕구불만도 = 0
TCVAR:ARG:파과 = 0

IF TALENT:ARG:연모 && CFLAG:MASTER:데이트중 != MAIN_MAP
	PRINTFORML 행위를 끝낸 %마스터는% 일어서려고 했지만, %조사처리(CALLNAME:ARG,"는")% 아직 함께 있어주길 바라는 것 같다……
	CALL ASK_YN("그래도 돌아간다","묵고 갈까")
	IF !RESULT
		GOTO BYEBYE
	ELSE
		CALL KOJO_MESSAGE_SEND("EVENT", 26, ARG, 6)
		;초대 플래그에 ARG를 대입
		CFLAG:ARG:초대 = ARG
		CFLAG:MASTER:초대 = ARG
		;초기위치를 OMANEKIBEYA()=98로 한다
		CFLAG:ARG:초기위치 = OMANEKIBEYA()
		;이동시킨다
		CFLAG:ARG:현재위치 = OMANEKIBEYA()
		CFLAG:MASTER:현재위치 = OMANEKIBEYA()
		IF NEMUKE() < 720 && BASE
			PRINTFORML %마스터는% 오늘 하루 %CALLNAME:ARG%의 방에서 보내기로 했다
			PRINTFORML 잠시 같이 자고, 일어나고, 청소하고, 는실난실 거리고, 식사를 만들어 먹고,
			PRINTFORMW 또 는실난실 거리고, 섹스하고 잤다
			VARSET 경험증가수
			IF 실전은어느쪽인가
				주입량 = ABL:ARG:애널 * MIN(ABL:ARG:친밀, 15) * 10
				경험증가수:21 += MIN(ABL:ARG:친밀, 15)
				경험증가수:17 += MIN(ABL:ARG:친밀, 15)
				경험증가수:7 += MIN(ABL:ARG:친밀, 15) * 2
				경험증가수:2 += MIN(ABL:ARG:친밀, 15) * 2
			ELSE
				주입량 = ABL:ARG:질 * MIN(ABL:ARG:친밀, 15) * 10
				경험증가수:20 += MIN(ABL:ARG:친밀, 15)
				경험증가수:16 += MIN(ABL:ARG:친밀, 15)
				경험증가수:6 += 2 * MIN(ABL:ARG:친밀, 15)
				경험증가수:1 += 2 * MIN(ABL:ARG:친밀, 15)
			ENDIF
			경험증가수:41 =  MIN(ABL:ARG:친밀, 15) * 2
			주입량 = MIN(주입량, BASE:MASTER:정력)
			CALL JUEL_UP_SPECIAL(LOCAL,"욕정","욕망", 15)
			IF !실전은어느쪽인가
				CALL ADD_SAMEN_V(ARG, 주입량)
				CALL JUEL_UP_SPECIAL(ARG,"쾌V","V감각", 10,1)
			ELSE
				CALL ADD_SAMEN_A(ARG, 주입량)
				CALL JUEL_UP_SPECIAL(ARG,"쾌A","A감각", 10,1)
			ENDIF
			FOR LOCAL, 0,58
				IF 경험증가수:LOCAL
					PRINTFORML %EXPNAME:LOCAL% ＋{경험증가수:LOCAL}
					EXP:ARG:LOCAL += 경험증가수:LOCAL
				ENDIF
			NEXT
			CALL ADD_EXP_LESNIAN(ABL:ARG:친밀 * 2, ARG, "PRINT")
			호감도상승량 = ABL:ARG:친밀 * 10
			EXP:MASTER:청소경험 ++
			EXP:MASTER:요리경험 ++
			TIME += 720 - NEMUKE()
		ELSE
			PRINTFORML 졸려졌으므로, 그대로 %조사처리(CALLNAME:ARG,"와")% 같이 잤다
			CALL ADD_EXP_LESNIAN(ABL:ARG:친밀, ARG, "PRINT")
			호감도상승량 = ABL:ARG:친밀 * 5
		ENDIF
		CALL CHANGE_CFLAG(2, ARG, 호감도상승량)
		CFLAG:MASTER:취침시간 = DAY * 1440 + TIME
		BEGIN AFTERTRAIN
		RETURN 1
	ENDIF
ENDIF

;TW당신은 정리하지 않고서는  견딜 수 없다
$BYEBYE
SIF GETBIT(TALENT:MASTER:성별, 1)
	DOWNBASE:MASTER:정력 = 주입량
PRINTFORM 뒤처리를 한 %마스터는%, \@ 실전은어느쪽인가? 항문으로# 비렬로\@부터 \@TCVAR:ARG:파과 ? 파과혈이 섞인# \@ 백탁이 넘쳐흐르고 있는
PRINTFORML \@ TALENT:ARG:회복속도? 이쪽을 상기된 표정으로 바라보고 있는#멍하니 있는 \@ %CALLNAME:ARG%에게,

IF ABL:ARG:정액중독 >= (3 + 실전은어느쪽인가 * 2)
	PRINTFORML 청소해줬으니까 대신 이쪽을 깨끗하게 해달라며 입가에 \@TCVAR:ARG:파과 ? 그녀의 순결을 빼앗은 지 얼마 안 된 #국물 투성이의 \@ 페니스를 들이댔다
	PRINTFORMW %조사처리(CALLNAME:ARG,"는")% \@ TALENT:ARG:회복속도 > 0 ? 황홀해 하는 표정으로 정성스럽게#멍한 표정으로 \@ 정액을 핥고 있다……
	SETCOLOR C_YELLOW
	PRINTFORML 정액경험 ＋{ABL:ARG:정액중독}
	EXP:ARG:정액경험 += ABL:ARG:정액중독
	RESETCOLOR
	CALL JUEL_UP_SPECIAL(ARG,"굴복","정액중독", 5,1)
	PRINTFORM 완전히 만족한 %마스터는%,
ENDIF

PRINTFORMW \@ CFLAG:MASTER:데이트중 == MAIN_MAP? 잘자# 너무 좋았다, 다음은 %GET_MAPNAME(MAIN_MAP)%에서 하자\@라고 하며 \@ CFLAG:MASTER:311 == CFLAG:ARG:311 ? 잠에 들었다# 떠났다\@
CFLAG:ARG:딸딸이들킴 = 2
IF !CFLAG:ARG:노팬티 && CFLAG:MASTER:311 != CFLAG:ARG:311 && CFLAG:MASTER:데이트중 != MAIN_MAP && PANTSNAME(EQUIP:ARG:하반신속옷2, ARG) != "노팬티"
	PRINTFORML 모처럼의 기회이니, 선물을 받아 가시겠습니까?
	CALL ASK_YN
	IF !RESULT
		PRINTFORML 돌아갈 때, %마스터는% %CALLNAME:ARG%의 %조사처리(PANTSNAME(EQUIP:ARG:하반신속옷2, ARG),"를")% 주워 품에 넣었다
		CALL PANTS_GET(ARG)
	ENDIF
ENDIF


code:
;-------------------------------------------------
;실행 판정
;-------------------------------------------------
@COM_ABLE366
SIF !TFLAG:100
	RETURN 0
SIF TARGET <= 0
	RETURN 0
;시간정지/수면/동침/우후후 안은 불가
SIF SHIRAHU(TARGET) != 1
	RETURN 0
;일괄관리
SIF GLOBAL_COMABLE(366)
	RETURN RESULT

SELECTCASE IRAI_CHARA_TO_PROGRESS(TARGET)
CASE "의뢰미수탁"
;빈슬롯이 없으면 불가
	SIF !SEARCH_IRAI_EMPTY()
		RETURN 0
;개개로 지정된 조건을 충족시키지 않을 때
	SIF !INT_DATA_IRAI(CFLAG:TARGET:의뢰내용, "의뢰 수탁 가능 조건", TARGET)
		RETURN 0
CASE "의뢰중"
;의뢰 확인/파기가 가능
CASE "의뢰 달성"
;성공 보고 가능
CASE "기한 마감", "보고 잊음", "의뢰 실패"
;실패 보고
CASE "의뢰 종료(성공)", "의뢰 종료(기한 마감)", "의뢰 종료(보고 잊음)", "의뢰 종료(실패)"
;여기는 의뢰 플래그의 소멸 기다려 어째서 손대지 않는다
		RETURN 0
CASEELSE
		RETURN 0
ENDSELECT
RETURN 1

@COM366
SELECTCASE IRAI_CHARA_TO_PROGRESS(TARGET)
CASE "의뢰미수탁"
	CALL IRAI_START(TARGET)
	RETURN RESULT
CASE "의뢰중"
	CALL IRAI_LOOK(TARGET)
	RETURN RESULT
CASE "의뢰 달성", "기한 마감", "보고 잊음", "의뢰 실패"
	CALL IRAI_REPORT(TARGET)
	RETURN 1
ENDSELECT

;-------------------------------------------------
;의뢰 수탁
;-------------------------------------------------
@IRAI_START(CHARA)
#DIM CHARA
;의뢰 제시시
PRINTFORMW 아무래도 %조사처리(CALLNAME:CHARA,"는")% %CALLNAME:MASTER%에게 부탁할 일이 있는 것 같다
CALL SHOW_IRAI_MESSAGE(CHARA, CFLAG:CHARA:의뢰내용, "의뢰 제시시")
CALL SHOW_IRAI_DATA(CFLAG:CHARA:의뢰내용, 0)
CALL ASK_YN("의뢰를 받는다", "거절한다")
IF RESULT
	PRINTFORML %마스터는% 정중하게 거절했다
	CALL SHOW_IRAI_MESSAGE(CHARA, CFLAG:CHARA:의뢰내용, "의뢰비수탁시")
	PRINTFORMW %조사처리(CALLNAME:CHARA,"는")% 유감스러운 얼굴을 하고 있다
	RETURN -1
ELSE
	CALL SET_IRAI_SLOT(CHARA)
	PRINTFORMW %마스터는% %CALLNAME:CHARA%의 의뢰를 받았다
	CALL SHOW_IRAI_MESSAGE(CHARA, CFLAG:CHARA:의뢰내용, "의뢰 수탁시")
	RETURN 1
ENDIF

;-------------------------------------------------
;의뢰 확인/파기
;-------------------------------------------------
@IRAI_LOOK(CHARA)
#DIM CHARA
#DIM CHK_SLOT
CHK_SLOT = SEARCH_IRAI_SLOT(CFLAG:CHARA:의뢰내용)
CALL SHOW_IRAI_MESSAGE(CHARA, CFLAG:CHARA:의뢰내용, "의뢰 확인시")
CALL SHOW_IRAI_DATA(CFLAG:CHARA:의뢰내용, CHK_SLOT)
CALL ASK_YN("확인", "파기")
SIF !RESULT
	RETURN -1
PRINTFORML 정말로?
CALL ASK_YN
SIF RESULT
	RETURN -1
CALL SHOW_IRAI_MESSAGE(CHARA, CFLAG:CHARA:의뢰내용, "의뢰 파기시")
PRINTFORMW 의뢰를 파기했습니다.
CALL CLEAR_IRAI_SLOT(CHARA, CHK_SLOT, "의뢰 종료(실패)")

;-------------------------------------------------
;의뢰 보고
;-------------------------------------------------
@IRAI_REPORT(CHARA)
#DIM CHARA
#DIM CHK_SLOT
CHK_SLOT = SEARCH_IRAI_SLOT(CFLAG:CHARA:의뢰내용)
;의뢰 성공시
;의뢰 실패시
SELECTCASE IRAI_CHARA_TO_PROGRESS(CHARA)
CASE "의뢰 달성"
	CALL SHOW_IRAI_MESSAGE(CHARA, CFLAG:CHARA:의뢰내용, "성공 보고시")
	CALL IRAI_CARRYOUT(CHARA, CFLAG:CHARA:의뢰내용, "의뢰 성공시")
CASE "기한 마감", "보고 잊음", "의뢰 실패"
	CALL SHOW_IRAI_MESSAGE(CHARA, CFLAG:CHARA:의뢰내용, "실패 보고시")
	CALL IRAI_CARRYOUT(CHARA, CFLAG:CHARA:의뢰내용, "의뢰 실패시")
ENDSELECT

;악보 입수 처리 ;KR판 악보기능 이식시 주석 해제
[SKIPSTART]
SELECTCASE IRAI_CHARA_TO_PROGRESS(CHARA)
CASE "의뢰 달성"
	RESULT = 楽譜入手_特殊条件(楽譜入手方法：妖精依頼成功)
	PRINTL
	SELECTCASE CHARA
	CASE 5,6,7,13,14
		RESULT = 楽譜入手_特殊条件(楽譜入手方法：妖精依頼成功)
	CASE 29
		RESULT = 楽譜入手_特殊条件(楽譜入手方法：文依頼成功)
	CASE 42
		RESULT = 楽譜入手_特殊条件(楽譜入手方法：はたて依頼成功)
	ENDSELECT
CASE "의뢰 실패"
	IF (RAND:(100) < 30)
		PRINTL
		RESULT = 楽譜入手_特殊条件(楽譜入手方法：依賴失敗時)
	ENDIF
ENDSELECT
[SKIPEND]
SELECTCASE IRAI_CHARA_TO_PROGRESS(CHARA)
CASE "의뢰 달성"
	CALL CLEAR_IRAI_SLOT(CHARA, CHK_SLOT, "의뢰 종료(성공)")
CASE "기한 마감"
	CALL CLEAR_IRAI_SLOT(CHARA, CHK_SLOT, "의뢰 종료(기한 마감)")
CASE "보고 잊음"
	CALL CLEAR_IRAI_SLOT(CHARA, CHK_SLOT, "의뢰 종료(보고 잊음)")
CASE "의뢰 실패"
	CALL CLEAR_IRAI_SLOT(CHARA, CHK_SLOT, "의뢰 종료(실패)")
ENDSELECT

path: ERB/커맨드관련/COMF/의뢰/COMF366 의뢰듣기_t.ERB


code:
;-------------------------------------------------
;실행 판정
;-------------------------------------------------
@COM_ABLE367
SIF !TFLAG:100
	RETURN 0
SIF !SEARCH_ACTIVE_IRAI_SLOT()
	RETURN 0
RETURN 1

@COM367
#DIM CHK_SLOT
CHK_SLOT  = SEARCH_ACTIVE_IRAI_SLOT()
CALL IRAI_APPLY(CHK_SLOT)
RETURN RESULT

;-------------------------------------------------
;의뢰실행 처리를 슬롯 지정으로 했다
;반드시 CAN_IRAI_SLOT_DO(CHK_SLOT)를 씹게 하고 나서 실행하지 않으면 소안내가 됩니다
;@COM367는 SEARCH_ACTIVE_IRAI_SLOT()로 판정이 끝난 상태이므로 좋아
;-------------------------------------------------
@IRAI_APPLY(CHK_SLOT)
#DIM CHK_SLOT
#DIM IRAI_RESULT
#DIM CLIENT_ID
CLIENT_ID = IRAI_SLOT_TO_CLIENT(CHK_SLOT)
CALL SHOW_IRAI_MESSAGE(TARGET, IRAI_SLOT:CHK_SLOT, "의뢰실행 직전")
CALL IRAI_CARRYOUT(TARGET, IRAI_SLOT:CHK_SLOT, "의뢰실행시")
IRAI_RESULT = RESULT
;여기서 설정하지 않으면 실행 직후의 메시지로 성공 실패의 판별이 되지 않기 때문에 어쩔 수 없이
SELECTCASE IRAI_RESULT
CASE IRAI_CONST_RESULT("캔슬")
	;여기는―1이 돌아간다
	;이외는 실행 처리를 했을 것이므로 모두 1이상을 돌려준다
	RETURN IRAI_RESULT
CASE IRAI_CONST_RESULT("실행해 성공")
	CALL SET_IRAI_PROGRESS(CLIENT_ID, "의뢰 달성")
CASE IRAI_CONST_RESULT("실행해 실패")
	CALL SET_IRAI_PROGRESS(CLIENT_ID, "의뢰 실패")
CASE IRAI_CONST_RESULT("실행해 미해결")
ENDSELECT
CALL SHOW_IRAI_MESSAGE(TARGET, IRAI_SLOT:CHK_SLOT, "의뢰실행 직후")

;우선 시간경과
SIF !FLAG:70
	TIME += 5

;성공 여부가 미해결이라면 패스
IF !GROUPMATCH(IRAI_RESULT, IRAI_CONST_RESULT("실행해 성공"), IRAI_CONST_RESULT("실행해 실패"))
;성공 여부가 확정해, TARGET가 의뢰자이다
ELSEIF TARGET == CLIENT_ID
	CALL COM366
;성공 여부가 확정해, TARGET가 의뢰자가 아니다
ELSE
	;의뢰 완료(보고하지 않으면 실패 취급)
	PRINTFORMW 잊지 말고 %CALLNAME:CLIENT_ID%에게 보고하자
ENDIF
RETURN 1

path: ERB/커맨드관련/COMF/의뢰/COMF367 의뢰실행_t.ERB


path: ERB/커맨드관련/COMF/의뢰/IRAI_10 니토리오이_t.ERB


path: ERB/커맨드관련/COMF/의뢰/IRAI_14 히나에게감주_t.ERB


path: ERB/커맨드관련/COMF/의뢰/IRAI_3 스이카에게술_t.ERB


code:
[SKIPSTART]
;--------------------------------------------
;기본 조건 여러 가지(5)
;--------------------------------------------
@IRAI_SETTING_4

IRAI_SETTING_DEADLINE:4 = 8
IRAI_SETTING_TARGET:4 = 29
IRAI_SETTING_CLIENT:4 = 29

IRAI_SETTING_COMMANDNAME:4 = 술을 건네준다

;--------------------------------------------
;의뢰 발생시의 메시지
;--------------------------------------------
@IRAI_INITIALMESSAGE_4
PRINTFORML 「술을 갖고 싶네요」
PRINTFORML 「어떤 것이라도 좋으니까, 3개(살) 정도 가져와 주세요」
PRINTFORML 「사례는 활기를 띠어요, 어때요?」
PRINTFORML 의뢰내용：술을 3개 이상 납입한다
PRINTFORML 보수　　：\\5000

;--------------------------------------------
;의뢰실행시의 메시지나 소스등의 처리
;--------------------------------------------
@IRAI_CARRYOUT_4
PRINTL 어떤 것을 건네줄까?
CALL 술일람표시(3)
IF RESULT >= 51 && RESULT <= 53
	PRINTFORMW %마스터는% %CALLNAME:TARGET%에 술을 건네주었다
	ITEM:RESULT -= 3
	;TFLAG에 건네준 술번호를 보존. 보수등을 분기 시키고 싶은 경우에.
	TFLAG:194 = RESULT
ELSE
	RETURN -1
ENDIF

;--------------------------------------------
;의뢰 실패시의 메시지나 소스등의 처리
;--------------------------------------------
@IRAI_FAILED_4
IF TALENT:29:연모
	PRINTFORML 「아야야, 조금 늦었습니다. 미안해요 저기, %CALLNAME:MASTER% 씨」
ELSE
	PRINTFORML 「죄송합니다만, 조금 늦었지요」
ENDIF
;시스템 메시지 다하고 호감도 변화
;마지막 인수만 호감도가 변화
CALL CHANGE_CFLAG(2, TARGET,-20)

;시스템 메시지 다하고 신뢰도 변화
;마지막 인수만 신뢰도가 변화
CALL CHANGE_CFLAG(4, TARGET,-10)
;--------------------------------------------
;의뢰 달성 때의 메시지나 소스등의 처리
;--------------------------------------------
@IRAI_SUCCESS_4
IF TALENT:29:연모
	PRINTFORML 「고마워요, %CALLNAME:MASTER% 씨♪」
ELSE
	PRINTFORML 「아, 감사합니다」
ENDIF
CALL KIGEN_CHANGE(TARGET, 100,1)
SOURCE:TARGET:환락 += 2000
;시스템 메시지 다하고 호감도 변화
;마지막 인수만 호감도가 변화
CALL CHANGE_CFLAG(2, TARGET, 30)

;시스템 메시지 다하고 신뢰도 변화
;마지막 인수만 신뢰도가 변화
CALL CHANGE_CFLAG(4, TARGET, 10)

MONEY += 5000
;--------------------------------------------
;의뢰실행 조건
;--------------------------------------------
@IRAI_ACTABLE_4
SIF SHIRAHU(TARGET) == 1 && (ITEM:51 >= 3 || ITEM:52 >= 3 || ITEM:53 >= 3)
	RETURN 1

;--------------------------------------------
;의뢰 발생 조건
;--------------------------------------------
@IRAI_HAPPEN_4
SIF !RAND:5
	RETURN 1

;--------------------------------------------
;의뢰 소멸 조건
;--------------------------------------------
@IRAI_VANISH_4
SIF !RAND:3
	RETURN 1
[SKIPEND]

path: ERB/커맨드관련/COMF/의뢰/IRAI_4 아야에게술_t.ERB


path: ERB/커맨드관련/COMF/의뢰/IRAI_6 대요정아이스_t.ERB


path: ERB/커맨드관련/COMF/의뢰/IRAI_8 메이린에게먹거리_t.ERB


path: ERB/커맨드관련/COMF/의뢰/IRAI_9 카리스마고갈_t.ERB